

The Agda Universal Algebra Library

Part 1: Foundation

Equality, extensionality, truncation, and dependent types for relations and algebras

William DeMeo   

Department of Algebra, Charles University in Prague

Abstract

The Agda Universal Algebra Library ([UALib](https://gitlab.com/ualib/ualib.gitlab.io)) is a library of types and programs (theorems and proofs) we developed to formalize the foundations of universal algebra in dependent type theory using the Agda programming language and proof assistant. The [UALib](https://gitlab.com/ualib/ualib.gitlab.io) includes a substantial collection of definitions, theorems, and proofs from general algebra and equational logic, including many examples that exhibit the power of inductive and dependent types for representing and reasoning about relations, algebraic structures, and equational theories. In this paper we describe several important aspects of the logical foundations on which the library is built. We also discuss (though sometimes only briefly) all of the types defined in the first 13 modules of the library, with special attention given to those details that seem most interesting or challenging from a type theory or mathematical foundations perspective.

2012 ACM Subject Classification Theory of computation → Logic and verification; Computing methodologies → Representation of mathematical objects; Theory of computation → Type theory

Keywords and phrases Agda, constructive mathematics, dependent types, equational logic, extensionality, formalization of mathematics, model theory, type theory, universal algebra

Related Version hosted on arXiv

Part 2, Part 3: http://arxiv.org/a/demeo_w_1

Supplementary Material

Documentation: ualib.org

Software: <https://gitlab.com/ualib/ualib.gitlab.io.git>

Acknowledgements

The author thanks [Martín Escardó](#) for creating the [Type Topology](#) library and teaching us about it at the [2019 Midlands Graduate School in Computing Science](#) [9].



This work and the Agda Universal Algebra Library by [William DeMeo](#) is licensed under a [Creative Commons Attribution-ShareAlike 4.0 International License](#).



© 2021 [William DeMeo](#). Based on work at <https://gitlab.com/ualib/ualib.gitlab.io>.
Compiled with [xelatex](#) on 18 Mar 2021 at 17:38.

Contents

1	Introduction	2
1.1	Motivation	2
1.2	Prior art	3
1.3	Attributions and Contributions	3
1.4	Organization of the paper	4
1.5	Resources	4
2	Agda Prelude	5
2.1	Preliminaries: logical foundations, universes, dependent types	5
2.2	Equality: definitional equality and transport	8
2.3	Extensionality: types for postulating function extensionality	11
2.4	Inverses: epics, monics, embeddings, inverse images	13
2.5	Lifts: making peace with a noncumulative universe hierarchy	16
3	Relation Types	17
3.1	Discrete: predicates, axiom of extensionality, compatibility	17
3.2	Continuous: arbitrary-sorted relations of arbitrary arity	20
3.3	Quotients: equivalences, class representatives, quotient types	22
3.4	Truncation: continuous propositions, quotient extensionality	24
4	Algebra Types	29
4.1	Signatures: types for operations and signatures	29
4.2	Algebras: types for algebras, operation interpretation, and compatibility	30
4.3	Products: types for products over arbitrary classes of algebras	32
4.4	Congruences: types for congruences and quotient algebras	34
5	Concluding Remarks	36

1 Introduction

To support formalization in type theory of research level mathematics in universal algebra and related fields, we present the Agda Universal Algebra Library ([AgdaUALib](#)), a software library containing formal statements and proofs of the core definitions and results of universal algebra. The [UALib](#) is written in [Agda](#) [12], a programming language and proof assistant based on Martin-Löf Type Theory that not only supports dependent and inductive types, as well as proof tactics for proving things about the objects that inhabit these types.

1.1 Motivation

The seminal idea for the [AgdaUALib](#) project was the observation that, on the one hand, a number of fundamental constructions in universal algebra can be defined recursively, and theorems about them proved by structural induction, while, on the other hand, inductive and dependent types make possible very precise formal representations of recursively defined objects, which often admit elegant constructive proofs of properties of such objects. An important feature of such proofs in type theory is that they are total functional programs and, as such, they are computable, composable, and machine-verifiable.

Finally, our own research experience has taught us that a proof assistant and programming language (like Agda), when equipped with specialized libraries and domain-specific tactics to

automate the proof idioms of our field, can be an extremely powerful and effective asset. As such we believe that proof assistants and their supporting libraries will eventually become indispensable tools in the working mathematician’s toolkit.

1.2 Prior art

There have been a number of efforts to formalize parts of universal algebra in type theory prior to ours, most notably

- Capretta [3] (1999) formalized the basics of universal algebra in the Calculus of Inductive Constructions using the Coq proof assistant;
- Spitters and van der Weegen [14] (2011) formalized the basics of universal algebra and some classical algebraic structures, also in the Calculus of Inductive Constructions using the Coq proof assistant, promoting the use of type classes as a preferable alternative to setoids;
- Gunther, et al [10] (2018) developed what seems to be (prior to the UALib) the most extensive library of formal universal algebra to date; in particular, this work includes a formalization of some basic equational logic; also (unlike the UALib) it handles *multisorted* algebraic structures; (like the UALib) it is based on dependent type theory and the Agda proof assistant.

Some other projects aimed at formalizing mathematics generally, and algebra in particular, have developed into very extensive libraries that include definitions, theorems, and proofs about algebraic structures, such as groups, rings, modules, etc. However, the goals of these efforts seem to be the formalization of special classical algebraic structures, as opposed to the general theory of (universal) algebras. Moreover, the part of universal algebra and equational logic formalized in the UALib extends beyond the scope of prior efforts and. In particular, the library now includes a proof of Birkhoff’s variety theorem. Most other proofs of this theorem that we know of are informal and nonconstructive.¹

1.3 Attributions and Contributions

The mathematical results described in this paper have well known *informal* proofs. Our main contribution is the formalization, mechanization, and verification of the statements and proofs of these results in dependent type theory using Agda.

Unless explicitly stated otherwise, the Agda source code described in this paper is due to the author, with the following caveat: the UALib depends on the [Type Topology](#) library of [Martín Escardó](#) [9]. Each dependency is carefully accounted for and mentioned in this paper. For the sake of completeness and clarity, and to keep the paper mostly self-contained, we repeat some definitions from the [Type Topology](#) library, but in each instance we cite the original source.²

In this paper we limit ourselves to the presentation of the core foundational modules of the UALib so that we have space to discuss some of the more interesting type theoretic and foundational issues that arose when developing the library and attempting to represent advanced mathematical notions in type theory and formalize them in Agda. As such, this is only the first installment of a three-part series of papers describing the [AgdaUALib](#). The second part is [7],

¹ After completing the formal proof in [Agda](#), we learned about a constructive version of Birkhoff’s theorem proved by Carlström in [4]. The latter is presented in the informal style of standard mathematical writing, and as far as we know it was never formalized in type theory and type-checked with a proof assistant. Nonetheless, a comparison of Carlström’s proof and the UALib proof would be interesting.

² In the UALib, such instances occur only inside hidden modules that are never actually used, followed immediately by a statement that imports the code in question from its original source.

covering homomorphisms, terms, and subalgebras. The third part is [8], which will cover free algebras, equational classes of algebras (i.e., varieties), and Birkhoff’s HSP theorem.

1.4 Organization of the paper

This present paper is organized into three parts as follows. The first part is §2 which introduces the basic concepts of type theory with special emphasis on the way such concepts are formalized in *Agda*. Specifically, §2.1 introduces *Sigma types* and *Agda*’s hierarchy of *universes*. The important topics of *equality* and *function extensionality* are discussed in §2.2 and §2.3; §2.4 covers inverses and inverse images of functions. In §2.5 we describe a technical problem that one frequently encounters when working in a *noncumulative universe hierarchy* and offer some tools for resolving the type-checking errors that arise from this.

The second part is §3 which covers *relation types* and *quotient types*. Specifically, §3.1 defines types that represent *unary* and *binary relations* as well as *function kernels*. These “discrete relation types,” are all very standard. In §3.2 we introduce the (less standard) types that we use to represent *general* and *dependent relations*. We call these “continuous relations” because they can have arbitrary arity (general relations) and they can be defined over arbitrary families of types (dependent relations). In §3.3 we cover standard types for equivalence relations and quotients, and in §3.4 we discuss a family of concepts that are vital to the mechanization of mathematics using type theory; these are the closely related concepts of *truncation*, *sets*, *propositions*, and *proposition extensionality*.

The third part of the paper is §4 which covers the basic domain-specific types offered by the *UALib*. It is here that we finally get to see some types representing algebraic structures. Specifically, we describe types for *operations* and *signatures* (§4.1), *general algebras* (§4.2), and *product algebras* (§4.3), including types for representing *products over arbitrary classes of algebraic structures*. Finally, we define types for congruence relations and quotient algebras in §4.4.

1.5 Resources

We conclude this introduction with some pointers to helpful reference materials. For the required background in Universal Algebra, we recommend the textbook by Clifford Bergman [1]. For the type theory background, we recommend the HoTT Book [13] and Escardó’s *Introduction to Univalent Foundations of Mathematics with Agda* [9].

The following are informed the development of the *UALib* and are highly recommended.

- *Introduction to Univalent Foundations of Mathematics with Agda*, Escardó [9].
- *Dependent Types at Work*, Bove and Dybjer [2].
- *Dependently Typed Programming in Agda*, Norell and Chapman [11].
- *Formalization of Universal Algebra in Agda*, Gunther, Gadea, Pagano [10].
- *Programming Languages Foundations in Agda*, Philip Wadler [18].

More information about *AgdaUALib* can be obtained from the following official sources.

- ualib.org (the web site) documents every line of code in the library.
- gitlab.com/ualib/ualib.gitlab.io (the source code) *AgdaUALib* is open source.³
- *The Agda UALib, Part 2: homomorphisms, terms, and subalgebras* [7].
- *The Agda UALib, Part 3: free algebras, equational classes, and Birkhoff’s theorem* [8].

³ License: [Creative Commons Attribution-ShareAlike 4.0 International License](https://creativecommons.org/licenses/by-sa/4.0/).

The first item links to the official [UALib](#) html documentation which includes complete proofs of every theorem we mention here, and much more, including the Agda modules covered in the first and third installments of this series of papers on the [UALib](#).

Finally, readers will get much more out of reading the paper if they download the [AgdaUALib](#) from <https://gitlab.com/ualib/ualib.gitlab.io>, install the library, and try it out for themselves.

2 Agda Prelude

2.1 Preliminaries: logical foundations, universes, dependent types

This section presents the [Prelude.Preliminaries](#) module of the [AgdaUALib](#), slightly abridged.⁴ This module defines (or imports) the most basic and important types of *Martin-Löf dependent type theory* (MLTT). Although this is standard, we take this opportunity to highlight aspects of the [UALib](#) syntax that may differ from that of “standard Agda.”

2.1.1 Logical foundations

The [UALib](#) is based on a minimal version of [MLTT](#) that is the same or very close to the type theory on which [Martín Escardó’s Type Topology](#) Agda library is based. We won’t go into great detail here because there are already other very nice resources available, such as the section [A spartan Martin-Löf type theory](#) of the lecture notes by [Escardó](#) just mentioned, the [ncatlab entry on Martin-Löf dependent type theory](#), as well as the [HoTT Book](#) [13].

We begin by noting that only a very small collection of objects is assumed at the jumping-off point for MLTT. We have the *primitive types* ([0](#), [1](#), and [N](#), denoting the empty type, one-element type, and natural numbers), the *type formers* ([+](#), [Π](#), [Σ](#), [Id](#), denoting *binary sum*, *product*, *sum*, and the *identity* type), and an infinite collection of *type universes* (types of types) and universe variables to denote them. Like Escardó’s, our universe variables are typically upper-case caligraphic letters from the latter half of the English alphabet (e.g., \mathcal{U} , \mathcal{V} , \mathcal{W} , etc.).

Specifying logical foundations in Agda

An Agda program typically begins by setting some options and by importing types from existing Agda libraries. Options are specified with the [OPTIONS](#) *pragma* and control the way Agda behaves by, for example, specifying the logical axioms and deduction rules we wish to assume when the program is type-checked to verify its correctness. Every Agda program in the [UALib](#) begins with the following line.

```
{-# OPTIONS -without-K -exact-split -safe #-}
```

 (1)

These options control certain foundational assumptions that Agda makes when type-checking the program to verify its correctness.

- `-without-K` disables [Streicher’s K axiom](#); see [15];
- `-exact-split` makes Agda accept only definitions that are *judgmental* equalities; see [17];
- `-safe` ensures that nothing is postulated outright—every non-MLTT axiom has to be an explicit assumption (e.g., an argument to a function or module); see [16] and [17].

Throughout this paper we take assumptions 1–3 for granted without mentioning them explicitly.

⁴ For unabridged docs and source code see <https://ualib.gitlab.io/Prelude.Preliminaries.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Prelude/Preliminaries.lagda>.

Agda Modules

The `OPTIONS` pragma is usually followed by the start of a module. For example, the `Prelude.Preliminaries` module begins with the following line.

```
module Prelude.Preliminaries where
```

Sometimes we want to declare parameters that will be assumed throughout the module. For instance, when working with algebras, we often assume they come from a particular fixed signature, and this signature is something we could fix as a parameter at the start of a module. Thus, we might start an *anonymous submodule* of the main module with a line like⁵

```
module _ {S : Signature} where
```

Such a module is called *anonymous* because an underscore appears in place of a module name. Agda determines where a submodule ends by indentation. This can take some getting used to, but after a short time it will feel very natural. The main module of a file must have the same name as the file, without the `.agda` or `.lagda` file extension. The code inside the main module is not indented. Submodules are declared inside the main module and code inside these submodules must be indented to a fixed column. As long as the code is indented, Agda considers it part of the submodule. A submodule is exited as soon as a nonindented line of code appears.

Agda's universe hierarchy

The `AgdaUALib` adopts the notation of Martin Escardo's `Type Topology` library [9]. In particular, *universe levels*⁶ are denoted by capitalized script letters from the second half of the alphabet, e.g., \mathcal{U} , \mathcal{V} , \mathcal{W} , etc. Also defined in `Type Topology` are the operators \cdot and $+$. These map a universe level \mathcal{U} to the universe $\mathcal{U} \cdot := \text{Set } \mathcal{U}$ and the level $\mathcal{U} + := \text{lsuc } \mathcal{U}$, respectively. Thus, $\mathcal{U} \cdot$ is simply an alias for the universe `Set \mathcal{U}` , and we have $\mathcal{U} \cdot : \mathcal{U} + \cdot$.

To justify the introduction of this somewhat nonstandard notation for universe levels, Escardó points out that the Agda library uses `Level` for universes (so what we write as $\mathcal{U} \cdot$ is written `Set \mathcal{U}` in standard Agda), but in univalent mathematics the types in $\mathcal{U} \cdot$ need not be sets, so the standard Agda notation can be a bit confusing, especially to new users.

The hierarchy of universes in Agda is structured as $\mathcal{U} \cdot : \mathcal{U} + \cdot$, $\mathcal{U} + \cdot : \mathcal{U} ++ \cdot$, etc. This means that the universe $\mathcal{U} \cdot$ has type $\mathcal{U} + \cdot$, and $\mathcal{U} + \cdot$ has type $\mathcal{U} ++ \cdot$, and so on. It is important to note, however, this does *not* imply that $\mathcal{U} \cdot : \mathcal{U} ++ \cdot$. In other words, Agda's universe hierarchy is *noncummulative*. This makes it possible to treat universe levels more generally and precisely, which is nice. On the other hand, a noncummulative hierarchy can sometimes make for a nonfun proof assistant. Luckily, there are ways to circumvent noncummulativity without introducing logical inconsistencies into the type theory. Section 4.2.4 describes some domain-specific tools that we developed for this purpose.

⁵ The `Signature` type will be defined in Section 4.1.

⁶ See <https://agda.readthedocs.io/en/v2.6.1.2/language/universe-levels.html>.

2.1.2 Dependent types

Sigma types (dependent pairs)

Given universes \mathcal{U} and \mathcal{V} , a type $X : \mathcal{U}^\bullet$, and a type family $Y : X \rightarrow \mathcal{V}^\bullet$, the *Sigma type* (aka *dependent pair type*, aka *dependent product type*) is denoted by $\Sigma(x : X), Y\ x$ and generalizes the Cartesian product $X \times Y$ by allowing the type $Y\ x$ of the second argument of the ordered pair (x, y) to depend on the value x of the first. That is, an inhabitant of the type $\Sigma(x : X), Y\ x$ is a pair (x, y) such that $x : X$ and $y : Y\ x$.

The dependent product type is defined in the [Type Topology](#) in a standard way. For pedagogical purposes we repeat the definition here.⁷

```
record  $\Sigma$  { $\mathcal{U} \ \mathcal{V}$ } { $X : \mathcal{U}^\bullet$ } ( $Y : X \rightarrow \mathcal{V}^\bullet$ ) :  $\mathcal{U} \sqcup \mathcal{V}^\bullet$  where
  constructor _,_
  field
    pr1 : X
    pr2 : Y pr1
```

Agda’s default syntax for this type is $\Sigma \lambda(x : X) \rightarrow Y$, but we prefer the notation $\Sigma\ x : X, Y$, which is closer to the standard syntax described in the preceding paragraph. Fortunately, the [Type Topology](#) library makes the preferred notation available with the following type definition and [syntax](#) declaration (see [9, Σ types]).⁸

```
- $\Sigma$  : { $\mathcal{U} \ \mathcal{V}$  : Universe} ( $X : \mathcal{U}^\bullet$ ) ( $Y : X \rightarrow \mathcal{V}^\bullet$ )  $\rightarrow \mathcal{U} \sqcup \mathcal{V}^\bullet$ 
- $\Sigma\ X\ Y = \Sigma\ X\ Y$ 

syntax - $\Sigma\ X\ (\lambda\ x \rightarrow Y) = \Sigma\ x : X, Y$ 
```

A special case of the Sigma type is the one in which the type Y doesn’t depend on X . This is the usual Cartesian product, defined in Agda as follows.

```
_×_ :  $\mathcal{U}^\bullet \rightarrow \mathcal{V}^\bullet \rightarrow \mathcal{U} \sqcup \mathcal{V}^\bullet$ 
 $X \times Y = \Sigma\ x : X, Y$ 
```

Pi types (dependent functions)

Given universes \mathcal{U} and \mathcal{V} , a type $X : \mathcal{U}^\bullet$, and a type family $Y : X \rightarrow \mathcal{V}^\bullet$, the *Pi type* (aka *dependent function type*) is denoted by $\Pi(x : X), Y\ x$ and generalizes the function type $X \rightarrow Y$ by letting the type $Y\ x$ of the codomain depend on the value x of the domain type. The dependent function type is defined in the [Type Topology](#) in a standard way. For the reader’s benefit, however, we repeat the definition here. (In the [UALib](#) this definition is included in a named or “hidden” module.)

```
 $\Pi$  : { $X : \mathcal{U}^\bullet$ } ( $A : X \rightarrow \mathcal{V}^\bullet$ )  $\rightarrow \mathcal{U} \sqcup \mathcal{V}^\bullet$ 
 $\Pi\ \{X\}\ A = (x : X) \rightarrow A\ x$ 
```

⁷ In the [UALib](#) we put such redundant definitions inside “hidden” modules so that they doesn’t conflict with the original definitions which we import and use. It may seem odd to define something in a hidden module only to import and use an alternative definition, but we do this in order to exhibit all of the types on which the [UALib](#) depends while ensuring that this cannot be misinterpreted as a claim to originality.

⁸ **Attention!** The symbol $:$ that appears in the special syntax defined here for the Σ type, and below for the Π type, is not the ordinary colon; rather, it is the symbol obtained by typing `\:4` in [agda2-mode](#).

To make the syntax for Π conform to the standard notation for Pi types, Escardó uses the same trick as the one used above for Sigma types.⁸

```
-Π : (X :  $\mathcal{U}$  ·) (Y : X →  $\mathcal{W}$  ·) →  $\mathcal{U}$   $\sqcup$   $\mathcal{W}$  ·
-Π X Y = Π Y
```

```
syntax -Π A (λ x → b) = Π x : A , b
```

Once we have studied the types, defined in the `Type Topology` library and repeated here for illustration purposes, the original definitions are imported like so.

```
open import Sigma-Type
open import MGS-MLTT using (pr1; pr2;  $\times$ ; -Σ; Π; -Π)
```

We use the `public` directive so that the types are available to all modules that import the present module.

2.1.2.1 Notation for the first and second projections

The definition of Σ (and thus \times) includes the fields `pr1` and `pr2` representing the first and second projections out of the product. Sometimes we prefer to denote these projections by `|_` and `||_`, respectively. However, for emphasis or readability we alternate between these and the following standard notations: `pr1` and `fst` for the first projection, `pr2` and `snd` for the second. We define these alternative notations for projections as follows.

```
module _ { $\mathcal{U}$  : Universe} where

|_ fst : {X :  $\mathcal{U}$  ·} {Y : X →  $\mathcal{V}$  ·} → Σ Y → X
| x , y | = x
fst (x , y) = x

||_ snd : {X :  $\mathcal{U}$  ·} {Y : X →  $\mathcal{V}$  ·} → (z : Σ Y) → Y (pr1 z)
|| x , y || = y
snd (x , y) = y
```

Note that we put the definitions above inside an *anonymous module*, which starts with the `module` keyword followed by an underscore (instead of a module name). The purpose is simply to move the postulated typing judgments—the “parameters” of the module (e.g., `\mathcal{U} : Universe`)—out of the way so they don’t obfuscate the definitions inside the module.

2.2 Equality: definitional equality and transport

This section presents the `Prelude.Equality` module of the `AgdaUALib`, slightly abridged.⁹ The type referred to as “reflexivity” or “refl” is a very basic but important type in `MLTT`. It represents *definitional equality*.¹⁰ The `refl` type we use is a standard one and is defined in the `Identity-Type` module of the `Type Topology` library. Apart from superficial syntactic differences it is equivalent to the identity type used in other Agda libraries.

⁹ For unabridged docs and source code see <https://ualib.gitlab.io/Prelude.Equality.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Prelude/Equality.lagda>.

¹⁰ For the definition and a discussion of definitional equality, see https://ncatlab.org/nlab/show/equality#definitional_equality.

In the `UALib` we make `refl` available by importing it from the `Identity-Type` module, but we repeat the definition here for easy reference.

```
data ==_ {U} {X : U → U} : X → X → U → where refl : {x : X} → x == x
```

Thus, whenever we need to complete a proof by simply asserting that x is definitionally equal to itself, we invoke `refl`. If we need to make x explicit, we use `refl {x = x}`.

Of course `==` is an equivalence relation and the formal proof of this is trivial. We don't need to prove reflexivity since it is the defining property of `==`. Here are the (trivial) proofs of symmetry and transitivity of `==`.

```
==-symmetric : (x y : X) → x == y → y == x
==-symmetric _ _ refl = refl

==-sym : {x y : X} → x == y → y == x
==-sym refl = refl

==-transitive : (x y z : X) → x == y → y == z → x == z
==-transitive _ _ _ refl refl = refl

==-trans : {x y z : X} → x == y → y == z → x == z
==-trans refl refl = refl
```

The only difference between `==-symmetric` and `==-sym` (respectively, `==-transitive` and `==-trans`) is that the latter has fewer explicit arguments, which is sometimes convenient.

Many proofs make abundant use of the symmetry of `==`, and the following syntactic sugar can improve the readability of such proofs.¹¹

```
_-1 : {X : U → U} → {x y : X} → x == y → y == x
p-1 = ==-sym p
```

If we have a proof $p : x == y$, and we need a proof of $y == x$, then instead of `==-sym p` we can use the more intuitive p^{-1} . Similarly, the following syntactic sugar makes abundant appeals to transitivity easier to stomach.

```
_·_ : {X : U → U} {x y z : X} → x == y → y == z → x == z
p · q = ==-trans p q
```

2.2.1 Transport

Alonzo Church characterized equality by declaring two things equal if and only if no property (predicate) can distinguish them (see [5]). In other terms, x and y are equal if and only if for all P we have $P x \rightarrow P y$. One direction of this implication is sometimes called *substitution* or *transport* or *transport along an identity*. It asserts the following: if two objects are equal and one of them satisfies a given predicate, then so does the other. A type representing this notion is defined, along with the (polymorphic) identity function, in the `MGS-MLTT` module of the `Type Topology` library, as follows.¹²

¹¹**Unicode Hints.** In `agda2-mode`, one types $^{-1}$ as `\^-1`, and `id` as `\MiiMid`, and type `·` as `\..`. In general, to get information about a given unicode character (e.g., how to type it) place the cursor over that character and type `M-x describe-char` (or `M-x h d c`).

¹²We can't show every line of code from the `UALib` in this paper, and we have chosen to omit the lines indicating that our redundant definitions of some functions (e.g., `transport` and `ap`) occur inside named

```

id : {X : Universe} (X : X → X) → X → X
id X = λ x → x

```

```

transport : {X : U → U} (A : X → W → W) {x y : X} → x ≡ y → A x → A y
transport A (refl {x = x}) = id (A x)

```

See [9] for a discussion of `transport`.¹³

A function is well defined if and only if it maps equivalent elements to a single element and we often use this nature of functions in Agda proofs. If we have a function $f : X \rightarrow Y$, two elements $x, x' : X$ of the domain, and an identity proof $p : x \equiv x'$, then we obtain a proof of $f x \equiv f x'$ by simply applying the `ap` function like so, `ap f p : f x ≡ f x'`. Escardó defines `ap` in the `Type Topology` library as follows.

```

ap : {X : U → U} {Y : V → V} (f : X → Y) {a b : X} → a ≡ b → f a ≡ f b
ap f {a} p = transport (λ - → f a ≡ f -) p (refl {x = f a})

```

Here are some variations of `ap` that are sometimes useful.

```

ap-cong : {A : U → U} {B : W → W} {f g : A → B} {a b : A} → f ≡ g → a ≡ b → f a ≡ g b
ap-cong refl refl = refl

```

We sometimes need a version of this that works for *dependent types*, such as the following (which we borrow from the `Relation/Binary/Core.agda` module of the `Agda Standard Library`, transcribed into our notation of course).

```

cong-app : {A : U → U} {B : A → W → W} {f g : Π B} → f ≡ g → (a : A) → f a ≡ g a
cong-app refl _ = refl

```

2.2.2 ≡-intro and ≡-elim for nondependent pairs

We conclude our presentation of the `Prelude.Equality` module with some occasionally useful introduction and elimination rules for the equality relation on (nondependent) pair types.

```

≡-elim-left : {A1 A2 : U → U} {B1 B2 : W → W} → (A1 , B1) ≡ (A2 , B2) → A1 ≡ A2
≡-elim-left e = ap pr1 e

≡-elim-right : {A1 A2 : U → U} {B1 B2 : W → W} → (A1 , B1) ≡ (A2 , B2) → B1 ≡ B2
≡-elim-right e = ap pr2 e

≡-×-intro : {A1 A2 : U → U} {B1 B2 : W → W} → A1 ≡ A2 → B1 ≡ B2 → (A1 , B1) ≡ (A2 , B2)
≡-×-intro refl refl = refl

≡-×-int : {A : U → U} {B : W → W} {a a' : A} {b b' : B} → a ≡ a' → b ≡ b' → (a , b) ≡ (a' , b')
≡-×-int refl refl = refl

```

(“hidden”) modules before the original definitions are then imported from the `Type Topology` library. As suggested above, we do this in order to facilitate presentation of the `UALib` library in a clear and reasonably self-contained way, without claiming credit for type definitions that are not our own.

¹³ cf. `transport` in `HoTT-Agda`: <https://github.com/HoTT/HoTT-Agda/blob/master/core/lib/Base.agda>.

2.3 Extensionality: types for postulating function extensionality

This section presents the `Prelude.Extensionality` module of the `AgdaUALib`, slightly abridged.¹⁴

2.3.1 Background and motivation

This brief introduction to the basics of *function extensionality* is intended for novices. If you're already familiar with the concept, you may want to skip to the next subsection.

What does it mean to say that two functions $f, g: X \rightarrow Y$ are equal? Suppose f and g are defined on $X = \mathbb{Z}$ (the integers) as follows: $fx := x + 2$ and $gx := ((2 * x) - 8)/2 + 6$. Would you say that f and g are equal? Are they the “same” function? What does that even mean?

If you know a little bit of basic algebra, then you probably can't resist the urge to reduce g to the form $x + 2$ and proclaim that f and g are, indeed, equal. And you would be right, at least in middle school, and the discussion would end there. In the science of computing, however, more attention is paid to equality, and with good reason.

We can probably all agree that the functions f and g above, while not syntactically equal, do produce the same output when given the same input so it seems fine to think of the functions as the same, for all intents and purposes. But we should ask ourselves, at what point do we notice or care about the difference in the way functions are defined?

What if we had started out this discussion with two functions f and g both of which take a list as input and produce as output a correctly sorted version of that list? Are the functions the same? What if f was defined using the `merge sort` algorithm, while g used `quick sort`? Probably most of us wouldn't think of f and g as the same function in that case.

In the examples above, it is common to say that the two functions f and g are *extensionally equal*, since they produce the same (external) output when given the same input, but they are not *intensionally equal*, since their (internal) definitions differ.

In this subsection, we describe types that manifest this idea of *extensional equality of functions*, or *function extensionality*. (Most of these types are already defined in the `Type Topology` library, so the `UALib` merely imports the definitions from there.)

2.3.2 Definition of function extensionality

As alluded to above, a natural notion of function equality, sometimes called *pointwise equality*, is defined as follows: f and g are said to be *pointwise equal* provided $\forall x \rightarrow fx \equiv gx$. Here is how this notion of equality is expressed as a type in the `Type Topology` library.

$$\begin{aligned} _ \sim _ &: \{ \mathcal{U} \ \mathcal{V} : \mathbf{Universe} \} \{ X : \mathcal{U} \ \cdot \} \{ A : X \rightarrow \mathcal{V} \ \cdot \} \rightarrow \Pi A \rightarrow \Pi A \rightarrow \mathcal{U} \sqcup \mathcal{V} \ \cdot \\ f \sim g &= \forall x \rightarrow fx \equiv gx \end{aligned}$$

Function extensionality is the assertion that pointwise equal functions are *definitionally equal*; that is, $\forall f \ g \ (f \sim g \rightarrow f \equiv g)$. In the `Type Topology` library, the types that represent this notion are `funext` (for nondependent functions) and `dfunext` (for dependent functions). They are defined as follows.

$$\begin{aligned} \text{funext} &: \forall \mathcal{U} \ \mathcal{W} \rightarrow (\mathcal{U} \sqcup \mathcal{W})^+ \ \cdot \\ \text{funext} \ \mathcal{U} \ \mathcal{W} &= \{ A : \mathcal{U} \ \cdot \} \{ B : \mathcal{W} \ \cdot \} \{ f \ g : A \rightarrow B \} \rightarrow f \sim g \rightarrow f \equiv g \end{aligned}$$

¹⁴For unabridged docs and source code see <https://ualib.gitlab.io/Prelude.Extensionality.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Prelude/Extensionality.lagda>.

```

dfunext :  $\forall \mathcal{U} \mathcal{W} \rightarrow (\mathcal{U} \sqcup \mathcal{W})^+ \cdot$ 
dfunext  $\mathcal{U} \mathcal{W} = \{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} \{f g : \forall (x : A) \rightarrow B\ x \rightarrow f \sim g \rightarrow f \equiv g$ 

```

In informal settings, this so-called “pointwise equality of functions” is typically what one means when one asserts that two functions are “equal.”¹⁵ However, it is important to keep in mind the following fact: *function extensionality is known to be neither provable nor disprovable in Martin-Löf type theory. It is an independent statement.* [9]

2.3.3 Global function extensionality

An assumption that we adopt throughout much of the current version of the `UALib` is a *global function extensionality principle*. This asserts that function extensionality holds at all universe levels. Agda is capable of expressing types representing global principles as the language has a special universe level for such types. Following Escardó, we denote this universe by `Uω` (which is just an alias for Agda’s `Setω` universe).¹⁶ The types `global-funext` and `global-dfunext` are defined in the `Type Topology` library as follows.

```

global-funext : Uω
global-funext =  $\forall \{ \mathcal{U} \mathcal{V} \} \rightarrow \text{funext } \mathcal{U} \mathcal{V}$ 

global-dfunext : Uω
global-dfunext =  $\forall \{ \mathcal{U} \mathcal{V} \} \rightarrow \text{funext } \mathcal{U} \mathcal{V}$ 

```

The next two types define the converse of function extensionality.¹⁷

```

extfun :  $\{A : \mathcal{U} \cdot\} \{B : \mathcal{W} \cdot\} \{f g : A \rightarrow B\} \rightarrow f \equiv g \rightarrow f \sim g$ 
extfun refl _ = refl

extdfun :  $\{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} (f g : \Pi B) \rightarrow f \equiv g \rightarrow f \sim g$ 
extdfun _ _ refl _ = refl

```

Though it may seem obvious to some readers, we wish to emphasize the important conceptual distinction between two different forms of type definitions we have seen so far. We do so by comparing the definitions of `funext` and `extfun`. In the definition of `funext`, the codomain is a generic type (namely, $(\mathcal{U} \sqcup \mathcal{V})^+ \cdot$). In the definition of `extfun`, the codomain is an assertion (namely, $f \sim g$). Also, the defining equation of `funext` is an assertion, while the defining equation of `extfun` is a proof. As such, `extfun` is a *proof object*; it proves (inhabits the type that represents) the proposition asserting that definitionally equivalent functions are pointwise equal. In contrast, `funext` is a type, and we may or may not wish to assume we have a proof for this type. That is, we could postulate that function extensionality holds and assume we have a witness, say, $fe : \text{funext } \mathcal{U} \mathcal{V}$ (i.e., a proof that pointwise equal functions are equal), but as noted above the existence of such a witness cannot be proved in Martin-Löf type theory.

¹⁵If one assumes the *univalence axiom* of Homotopy Type Theory, then the `_~_` relation is equivalent to equality of functions. See the section “[Function extensionality from univalence](#)” of Escardó’s notes [9].

¹⁶More details about the `Uω` type are available at agda.readthedocs.io.

¹⁷In previous versions of the `UALib` this function was called `intensionality`, indicating that it represented the concept of *function intensionality*, but we realized this isn’t quite right and changed the name to the less controversial `extfun`. Later we realized that a function called `happly`, which is nearly identical to `extdfun`, is already defined in the `MGS-FunExt-from-Univalence` module of the `Type Topology` library.

2.3.4 Alternative extensionality type

Finally, a useful alternative for expressing dependent function extensionality, which is essentially equivalent to `dfunext`, is to assert that `extdfun` is actually an *equivalence*. This requires a few definitions from the `MGS-Equivalences` module of the `Type Topology` library, which we now describe.

First, a type is a *singleton* if it has exactly one inhabitant and a *subsingleton* if it has at most one inhabitant. These properties are represented in the `Type Topology` library by the following type.

```
is-center : (X :  $\mathcal{U}$  ·) → X →  $\mathcal{U}$  ·
is-center X c = (x : X) → c ≡ x

is-singleton :  $\mathcal{U}$  · →  $\mathcal{U}$  ·
is-singleton X =  $\Sigma$  c : X , is-center X c

is-subsingleton :  $\mathcal{U}$  · →  $\mathcal{U}$  ·
is-subsingleton X = (x y : X) → x ≡ y
```

Next, we consider the type `is-equiv` which is used to assert that a function is an equivalence in the sense that we now describe. First we need the concept of a *fiber* of a function. In the `Type Topology` library, `fiber` is defined as a Sigma type whose inhabitants represent inverse images of points in the codomain of the given function.

```
fiber : {X :  $\mathcal{U}$  ·} {Y :  $\mathcal{W}$  ·} (f : X → Y) → Y →  $\mathcal{U} \sqcup \mathcal{W}$  ·
fiber {X} f y =  $\Sigma$  x : X , f x ≡ y
```

A function is called an *equivalence* if all of its fibers are singletons.

```
is-equiv : {X :  $\mathcal{U}$  ·} {Y :  $\mathcal{W}$  ·} → (X → Y) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
is-equiv f =  $\forall$  y → is-singleton (fiber f y)
```

Now we are finally ready to define the type `hfunext` that gives an alternative means of postulating function extensionality.

```
hfunext : ( $\mathcal{U} \mathcal{W}$  : Universe) → ( $\mathcal{U} \sqcup \mathcal{W}$ )+ ·
hfunext  $\mathcal{U} \mathcal{W}$  = {A :  $\mathcal{U}$  ·} {B : A →  $\mathcal{W}$  ·} (f g :  $\Pi$  B) → is-equiv (extdfun f g)
```

2.4 Inverses: epics, monics, embeddings, inverse images

This section presents the `Prelude.Inverses` module of the `AgdaUALib`, slightly abridged.¹⁸ We begin by defining an inductive type that represents the semantic concept of *inverse image* of a function.

```
data Image_⊃_ : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) : B →  $\mathcal{U} \sqcup \mathcal{W}$  ·
where
  im : (x : A) → Image f ⊃ f x
  eq : (b : B) → (a : A) → b ≡ f a → Image f ⊃ b
```

Next we verify that the type just defined is what we expect.

¹⁸For unabridged docs and source code see <https://ualib.gitlab.io/Prelude.Inverses.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Prelude/Inverses.lagda>.

```
ImageIsImage : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) (b : B) (a : A)
```

```
→      b ≡ f a → Image f ∋ b
```

```
ImageIsImage f b a b ≡ f a = eq b a b ≡ f a
```

Note that an inhabitant of `Image f ∋ b` is a pair (a, p) , where $a : A$, and p is a proof that f maps a to b ; that is, $p : b \equiv f a$. Since the proof that b belongs to the image of f is always accompanied by a “witness” $a : A$, we can actually *compute* a *pseudoinverse* of f . This function takes an arbitrary $b : B$ and a *(witness, proof)*-pair, $(a, p) : \text{Image } f \ni b$, and returns a .

```
Inv : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) {b : B} → Image f ∋ b → A
```

```
Inv f {.(f a)} (im a) = a
```

```
Inv f (eq _ a _) = a
```

We can prove that `Inv f` is the *right-inverse* of f , as follows.

```
InvIsInv : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) {b : B} (q : Image f ∋ b) → f (Inv f q) ≡ b
```

```
InvIsInv f {.(f a)} (im a) = refl
```

```
InvIsInv f (eq _ _ p) = p-1
```

2.4.1 Surjective functions

An *epic* (or *surjective*) function from type $A : \mathcal{U} \cdot$ to type $B : \mathcal{W} \cdot$ is as an inhabitant of the `Epic` type, which we define as follows.

```
Epic : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (g : A → B) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
```

```
Epic g =  $\forall y \rightarrow \text{Image } g \ni y$ 
```

We obtain the right-inverse (or pseudoinverse) of an epic function f by applying the function `EpicInv` (which we now define) to the function f along with a proof, $fepi : \text{Epic } f$, that f is surjective.

```
EpicInv : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) → Epic f → B → A
```

```
EpicInv f fepi b = Inv f (fepi b)
```

The function defined by `EpicInv f fepi` is indeed the right-inverse of f . To state this, we’ll use the function composition operation `◦`, which is already defined in the `Type Topology` library, as follows.

```
__◦__ : {X :  $\mathcal{U}$  ·} {Y :  $\mathcal{W}$  ·} {Z :  $\mathcal{W}$  ·} (f : X → Y) (g : Y → Z) → X → Z
```

```
g ◦ f =  $\lambda x \rightarrow g (f x)$ 
```

```
EpicInvIsRightInv : funext  $\mathcal{W} \mathcal{W} \rightarrow \{A : \mathcal{U} \cdot\} \{B : \mathcal{W} \cdot\}$ 
```

```
(f : A → B) (fepi : Epic f)
```

```
→      f ◦ (EpicInv f fepi) ≡ id B
```

```
EpicInvIsRightInv fe f fepi = fe ( $\lambda x \rightarrow \text{InvIsInv } f (fepi x)$ )
```

2.4.2 Injective functions

We say that a function $g : A \rightarrow B$ is **monic** (or **injective** or **one-to-one**) if it doesn’t map distinct elements to a common point. This property is formalized quite naturally using the

Monic type, which we now define.

```
Monic : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (g : A → B) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
Monic g =  $\forall$  a1 a2 → g a1  $\equiv$  g a2 → a1  $\equiv$  a2
```

Again, we obtain a pseudoinverse. Here it is obtained by applying the function **MonicInv** to *g* and a proof that *g* is monic.

```
MonicInv : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) → Monic f → (b : B) → Image f  $\ni$  b → A
MonicInv f _ =  $\lambda$  b Imf  $\ni$  b → Inv f Imf  $\ni$  b
```

The function defined by **MonicInv** *f* *M* is the left-inverse of *f*.

```
MonicInvIsLeftInv : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{W}$  ·} (f : A → B) (fmonic : Monic f) (x : A)
→ (MonicInv f fmonic) (f x) (im x)  $\equiv$  x
MonicInvIsLeftInv f fmonic x = refl
```

2.4.3 Embeddings

The type **is-embedding** *f* denotes the assertion that *f* is a function all of whose fibers are sub-singletons.

```
is-embedding : {X :  $\mathcal{U}$  ·} {Y :  $\mathcal{W}$  ·} → (X → Y) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
is-embedding f =  $\forall$  y → is-subsingleton (fiber f y)
```

This is a natural way to represent what we usually mean in mathematics by embedding. Observe that an embedding does not simply correspond to an injective map. However, if we assume that the codomain *B* has unique identity proofs (i.e., *B* is a *set*), then we can prove that a monic function into *B* is an embedding. We postpone this until we arrive at the **Relations.Truncation** module and take up the topic of sets.

It should be clear that embeddings are monic; from a proof *p* : **is-embedding** *f* that *f* is an embedding we can construct a proof of **Monic** *f*. We verify this as follows.

```
embedding-is-monic : { $\mathcal{X}$   $\mathcal{Y}$  : Universe} {X :  $\mathcal{X}$  ·} {Y :  $\mathcal{Y}$  ·}
(f : X → Y) → is-embedding f → Monic f
```

```
embedding-is-monic f femb x x' fxfx' = ap pr1 ((femb (f x)) fa fb)
```

where

```
fa : fiber f (f x)
```

```
fa = x , refl
```

```
fb : fiber f (f x)
```

```
fb = x' , (fxfx' -1)
```

Finally, one way to show that a function is an embedding is to first prove it is invertible and then invoke the following theorem.

```
invertibles-are-embeddings : { $\mathcal{X}$   $\mathcal{Y}$  : Universe} {X :  $\mathcal{X}$  ·} {Y :  $\mathcal{Y}$  ·} (f : X → Y)
→ invertible f → is-embedding f
```

```
invertibles-are-embeddings f fi = equivs-are-embeddings f (invertibles-are-equivs f fi)
```

2.5 Lifts: making peace with a noncumulative universe hierarchy

This section presents the `Prelude.Lifts` module of the `AgdaUALib`, slightly abridged.¹⁹

2.5.1 The noncumulative universe hierarchy

The hierarchy of universe levels in Agda looks like this:

$$\mathcal{U}_0 : \mathcal{U}_1 \cdot, \quad \mathcal{U}_1 : \mathcal{U}_2 \cdot, \quad \mathcal{U}_2 : \mathcal{U}_3 \cdot, \dots$$

This means that the type level of \mathcal{U}_0 is $\mathcal{U}_1 \cdot$, and for each n the type level of \mathcal{U}_n is $\mathcal{U}_{n+1} \cdot$. It is important to note, however, this does *not* imply that $\mathcal{U}_0 : \mathcal{U}_2 \cdot$ and $\mathcal{U}_0 : \mathcal{U}_3 \cdot$, and so on. In other words, Agda’s universe hierarchy is *noncumulative*. This makes it possible to treat universe levels more generally and precisely, which is nice. On the other hand (in this author’s experience) a noncumulative hierarchy can sometimes make for a nonfun proof assistant.

Luckily, there are ways to overcome this technical issue. We describe general lifting and lowering functions below, and then later, in §4.2.4, we’ll see the domain-specific analogs of these tools which turn out to have some nice properties that make them very effective for resolving universe level problems when working with algebra types.

2.5.2 Lifting and lowering

Let us be more concrete about what is at issue here by giving an example. Agda frequently encounters errors during the type-checking process and responds by printing an error message. Often the message has the following form.

```
Algebras.lagda:498,20-23  $\mathcal{U} \neq \mathbb{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$  when checking that... has type...
```

This error message means that Agda encountered the universe \mathcal{U} on line 498 (columns 20–23) of the file `Algebras.lagda`, but was expecting to find the universe $\mathbb{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$ instead.

To make these situations easier to deal with, we have developed some domain specific tools for the lifting and lowering of universe levels inhabited by some of the key algebraic types of the `UALib`. These tools must be applied with some care to avoid making the type theory inconsistent. In particular, we cannot lower the level of a type unless it was previously lifted to a (higher than necessary) universe level.

A general `Lift` record type, similar to the one found in the `Level` module of the `Agda Standard Library`, is defined as follows.

```
record Lift { $\mathcal{W} \mathcal{U} : \text{Universe}$ } ( $X : \mathcal{U} \cdot$ ) :  $\mathcal{U} \sqcup \mathcal{W} \cdot$  where
  constructor lift
  field lower :  $X$ 
open Lift
```

The point of having a ramified hierarchy of universes is to avoid Russell’s paradox, and this would be subverted if we were to lower the universe of a type that wasn’t previously lifted. However, we can prove that if an application of `lower` is immediately followed by an application of `lift`, then the result is the identity transformation. Similarly, `lift` followed by `lower` is the identity.

```
lower~lift : { $\mathcal{W} \mathcal{X} : \text{Universe}$ } { $X : \mathcal{X} \cdot$ } → lower{ $\mathcal{W}$ }{ $\mathcal{X}$ } ∘ lift ≡ id X
lower~lift = refl
```

¹⁹For unabridged docs and source code see <https://ualib.gitlab.io/Prelude.Lifts.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Prelude/Lifts.lagda>.


```
lift~lower : {W X : Universe} {X : X →} → lift ∘ lower ≡ id (Lift{W}{X} X)
lift~lower = refl
```

Evidently, the proofs are trivial. Nonetheless, we'll find a few holes that these observations can fill.

3 Relation Types

Here we present the submodules of the AgdaUALib's `Relations` module. In §3.1 we cover *unary* and *binary relations*, which we refer to as “discrete” to contrast them with the (“continuous”) *general* and *dependent relations* that we take up in §3.2. We call the latter “continuous relations” because they can have arbitrary arity (general relations) and they can be defined over arbitrary families of types (dependent relations).

3.1 Discrete: predicates, axiom of extensionality, compatibility

This section presents the `Relations.Discrete` module of the AgdaUALib, slightly abridged.²⁰

3.1.1 Unary relations

We need a mechanism for implementing the notion of subsets in Agda. A typical one is called `Pred` (for predicate). More generally, `Pred A U` can be viewed as the type of a property that elements of type `A` might satisfy. We write $P : \text{Pred } A \ U$ to represent the semantic concept of a collection of elements of type `A` that satisfy the property P . Here is the definition (which is similar to the one found in the `Relation/Unary.agda` file of the Agda Standard Library).

```
Pred : U → (W : Universe) → U → W →
Pred A W = A → W →
```

If we are given a type $A : U$, then we think of `Pred A W` as the type of a property that inhabitants of `A` may or may not satisfy. If $P : \text{Pred } A \ W$, then we can view P as a collection of inhabitants of type `A` that “satisfy property P ,” or that “belong to the subset P of `A`.”

Below we will often consider predicates over the class of all algebras of a particular type. Soon we will define the type `Algebra U S` of algebras (for some universe level `U`), and the type `Pred (Algebra U S) U` will be inhabited by maps of the form $\mathbf{A} \rightarrow U$; more precisely, given an algebra $\mathbf{A} : \text{Algebra } U \ S$, we will consider the type `Pred A U = A → U`. We will use predicates over algebra types to specify subclasses of algebras with certain properties.

3.1.2 Membership and inclusion relations

Of course, the UALib includes types that represent element-wise inclusion and subset containment. For example, given a predicate `P`, we may represent that “ x belongs to `P`” or that “ x has property `P`,” by writing either $x \in P$ or $P \ x$. The definition of `∈` is standard (cf. `Relation/Unary.agda` in the Agda Standard Library). Nonetheless, here it is.

```
_∈_ : {A : X →} → A → Pred A Y → Y
x ∈ P = P x
```

²⁰For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Discrete.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Discrete.lagda>.

The *subset* relation is denoted, as usual, with the \subseteq symbol and is defined as follows.

$$\begin{aligned} _ \subseteq _ &: \{A : \mathcal{X} \cdot\} \rightarrow \text{Pred } A \mathcal{Y} \rightarrow \text{Pred } A \mathcal{X} \rightarrow \mathcal{X} \sqcup \mathcal{Y} \sqcup \mathcal{X} \cdot \\ P \subseteq Q &= \forall \{x\} \rightarrow x \in P \rightarrow x \in Q \end{aligned}$$

3.1.3 The axiom of extensionality

In type theory everything is represented as a type and, as we have just seen, this includes subsets. Equality of types is a nontrivial matter, and thus so is equality of subsets when represented as unary predicates. Fortunately, it is straightforward to write down a type that represents what it typically means in informal mathematics to say that two subsets are (extensionally) equal—namely, they contain the same elements. In the `UALib` we denote this type by $\dot{=}$ and define it as follows.²¹

$$\begin{aligned} _ \dot{=} _ &: \{\mathcal{X} \mathcal{Y} \mathcal{Z} : \text{Universe}\} \{A : \mathcal{X} \cdot\} \rightarrow \text{Pred } A \mathcal{Y} \rightarrow \text{Pred } A \mathcal{Z} \rightarrow \mathcal{X} \sqcup \mathcal{Y} \sqcup \mathcal{Z} \cdot \\ P \dot{=} Q &= (P \subseteq Q) \times (Q \subseteq P) \end{aligned}$$

A proof of $P \dot{=} Q$ is a pair (p, q) where p is a proof of the first inclusion (that is, $p : P \subseteq Q$), and q is a proof of the second.

If P and Q are definitionally equal (i.e., $P \equiv Q$), then of course both $P \subseteq Q$ and $Q \subseteq P$ will hold, so $P \dot{=} Q$ will also hold.

$$\begin{aligned} \text{Pred-}\equiv &: \{\mathcal{X} \mathcal{Y} : \text{Universe}\} \{A : \mathcal{X} \cdot\} \{P Q : \text{Pred } A \mathcal{Y}\} \rightarrow P \equiv Q \rightarrow P \dot{=} Q \\ \text{Pred-}\equiv \text{ refl} &= (\lambda z \rightarrow z), (\lambda z \rightarrow z) \end{aligned}$$

The converse of `Pred- \equiv` is not provable in Martin-Löf Type Theory. However, we can postulate it axiomatically if we wish. This is called the *axiom of extensionality* and a type that represents it can be defined in `Agda` as follows.

$$\begin{aligned} \text{ext-axiom} &: \{\mathcal{X} : \text{Universe}\} \rightarrow \mathcal{X} \cdot \rightarrow (\mathcal{Y} : \text{Universe}) \rightarrow \mathcal{X} \sqcup \mathcal{Y} \cdot \\ \text{ext-axiom } A \mathcal{Y} &= \forall (P Q : \text{Pred } A \mathcal{Y}) \rightarrow P \dot{=} Q \rightarrow P \equiv Q \end{aligned}$$

We treat extensionality in greater detail and generality in §2.3 and §3.4.

3.1.4 Binary Relations

In set theory, a binary relation on a set A is simply a subset of the Cartesian product $A \times A$. As such, we could model these as predicates over the product type $A \times A$, or as relations of type $A \rightarrow A \rightarrow \mathcal{R}$ (for some universe \mathcal{R}). We define these below.

A generalization of the notion of binary relation is a *relation from A to B* , which we define first and treat binary relations on a single A as a special case.

$$\begin{aligned} \text{REL} &: \mathcal{U} \cdot \rightarrow \mathcal{R} \cdot \rightarrow (\mathcal{N} : \text{Universe}) \rightarrow (\mathcal{U} \sqcup \mathcal{R} \sqcup \mathcal{N} \cdot) \cdot \\ \text{REL } A B \mathcal{N} &= A \rightarrow B \rightarrow \mathcal{N} \cdot \end{aligned}$$

$$\begin{aligned} \text{Rel} &: \mathcal{U} \cdot \rightarrow (\mathcal{N} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{N} \cdot \\ \text{Rel } A \mathcal{N} &= \text{REL } A A \mathcal{N} \end{aligned}$$

The *kernel* of $f : A \rightarrow B$ is defined informally by $\{(x, y) \in A \times A : f x = f y\}$. This can be represented in type theory and `Agda` in a number of ways, each of which may be useful in

²¹**Unicode Hint.** In `agda2-mode` type `\doteq` or `\.=` to produce $\dot{=}$

a particular context. For example, we could define the kernel as a Sigma type (omitted), or as a unary relation (predicate) over the Cartesian product $A \times A$,

```
KER-pred : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{R}$  ·} → (A → B) → Pred (A × A)  $\mathcal{R}$ 
KER-pred g (x , y) = g x ≡ g y
```

or as a relation from A to A ,

```
KER-rel : {A :  $\mathcal{U}$  ·} {B :  $\mathcal{R}$  ·} → (A → B) → Rel A  $\mathcal{R}$ 
KER-rel g x y = g x ≡ g y
```

For example, the *identity relation* (the kernel of an injective function) could be represented as a Sigma type (omitted), or a relation, or a predicate.

```
0-rel : {A :  $\mathcal{U}$  ·} → Rel A  $\mathcal{U}$ 
0-rel a b = a ≡ b

0-pred : {A :  $\mathcal{U}$  ·} → Pred (A × A)  $\mathcal{U}$ 
0-pred (a , a') = a ≡ a'
```

3.1.5 The implication relation

We denote and define *implication* for binary predicates (relations) as follows.

```
_on_ : { $\mathcal{X}$   $\mathcal{Y}$   $\mathcal{Z}$  : Universe} {A :  $\mathcal{X}$  ·} {B :  $\mathcal{Y}$  ·} {C :  $\mathcal{Z}$  ·}
→ (B → B → C) → (A → B) → (A → A → C)

R on g = λ x y → R (g x) (g y)
```

```
_⇒_ : { $\mathcal{W}$   $\mathcal{X}$   $\mathcal{Y}$   $\mathcal{Z}$  : Universe} {A :  $\mathcal{W}$  ·} {B :  $\mathcal{X}$  ·}
→ Rel A B  $\mathcal{Y}$  → Rel A B  $\mathcal{Z}$  →  $\mathcal{W} \sqcup \mathcal{X} \sqcup \mathcal{Y} \sqcup \mathcal{Z}$  ·
```

```
P ⇒ Q = ∀ {i j} → P i j → Q i j
```

We can combine `_on_` and `_⇒_` to define a nice, general implication operation. (This is borrowed from the [Agda Standard Library](#); we have merely translated into [Type Topology/UALib](#) notation.)

```
_=[_]⇒_ : { $\mathcal{W}$   $\mathcal{X}$   $\mathcal{Y}$   $\mathcal{Z}$  : Universe} {A :  $\mathcal{W}$  ·} {B :  $\mathcal{X}$  ·}
→ Rel A  $\mathcal{Y}$  → (A → B) → Rel B  $\mathcal{Z}$  →  $\mathcal{W} \sqcup \mathcal{Y} \sqcup \mathcal{Z}$  ·
```

```
P =[ g ]⇒ Q = P ⇒ (Q on g)
```

3.1.6 Compatibility of functions and binary relations

Before discussing general and dependent relations, we pause to define some types that are useful for asserting and proving facts about *compatibility* of functions with binary relations. The first definition simply lifts a binary relation on A to a binary relation on tuples of type $I \rightarrow A$.²²

```
module _ { $\mathcal{U}$   $\mathcal{V}$   $\mathcal{W}$  : Universe} {I :  $\mathcal{V}$  ·} {A :  $\mathcal{U}$  ·} where
```

²²N.B. This *relation* lifting is not to be confused with the sort of *universe* lifting that we defined in the [Prelude.Lifts](#) module.

```
lift-rel : Rel A  $\mathcal{W}$  → (I → A) → (I → A) →  $\mathcal{V} \sqcup \mathcal{W}$  ·
lift-rel R a a' =  $\forall i \rightarrow R (a\ i) (a'\ i)$ 
```

```
compatible-fun : (f : (I → A) → A)(R : Rel A  $\mathcal{W}$ ) →  $\mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}$  ·
compatible-fun f R = (lift-rel R) = $[f] \Rightarrow$  R
```

We used the slick implication notation in the definition of `compatible-fun`, but we could have defined it more explicitly, like so.

```
compatible-fun' : (f : (I → A) → A)(R : Rel A  $\mathcal{W}$ ) →  $\mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}$  ·
compatible-fun' f R =  $\forall a\ a' \rightarrow (\text{lift-rel } R)\ a\ a' \rightarrow R\ (f\ a)\ (f\ a')$ 
```

However, this is a rare case in which the more elegant syntax may result in simpler proofs when applying the definition. (See, for example, `compatible-term` in the `Terms.Operations` module.)

3.2 Continuous: arbitrary-sorted relations of arbitrary arity

This section presents the `Relations.Continuous` module of the `AgdaUALib`, slightly abridged.²³ In set theory, an n -ary relation on a set A is simply a subset of the n -fold product $A \times A \times \cdots \times A$. As such, we could model these as predicates over the type $A \times \cdots \times A$, or as relations of type $A \rightarrow A \rightarrow \cdots \rightarrow A \rightarrow \mathcal{W}$ · (for some universe \mathcal{W}). To implement such a relation in type theory, we would need to know the arity in advance, and then somehow form an n -fold arrow \rightarrow .

A more general and straightforward approach is to instead define an arity type $I : \mathcal{V}$ ·, and define the type representing I -ary relations on A as the function type $(I \rightarrow A) \rightarrow \mathcal{W}$ ·. Then, if we are specifically interested in an n -ary relation for some natural number n , we could take I to be a finite set (e.g., of type `Fin n`).

Below we will define `ConRel` to be the type $(I \rightarrow A) \rightarrow \mathcal{W}$ · and we will call `ConRel` the type of *continuous relations*. This generalizes the discrete relations we defined in `[Relations.Discrete]` (unary, binary, ternary, etc.) since continuous relations can be of arbitrary arity. Still, they are not completely general since they are defined over a single type—said another way, these are “single-sorted” relations—but we will remove this limitation as well when we define the type of **dependent continuous relations**.

Just as `Rel A \mathcal{W}` · was the single-sorted special case of the multisorted `REL A B \mathcal{W}` · type, so too will `ConRel I A \mathcal{W}` · be the single-sorted version of a completely general type of relations. The latter will represent relations that not only have arbitrary arities, but also are defined over arbitrary families of types.

To be more concrete, given an arbitrary family $A : I \rightarrow \mathcal{U}$ · of types, we may have a relation from $A\ i$ to $A\ j$ to $A\ k$ to \dots , *ad infinitum*, where the collection represented by the “indexing” type I might not even be enumerable.²⁴

We will refer to such relations as *dependent continuous relations* (or *dependent relations*) because the definition of a type that represents them requires dependent types. The `DepRel` type that we define below manifests this completely general notion of relation.

²³For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Continuous.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Continuous.lagda>.

²⁴Because the collection represented by the indexing type I might not even be enumerable, technically speaking, instead of “ $A\ i$ to $A\ j$ to $A\ k$ to \dots ,” we should have written something like “`TO (i : I) , A i`”

3.2.1 Continuous relation types

We now define the type *ConRel* that represents predicates (or relations) of arbitrary arity over a single type *A*. We call this the type of **continuous relations**.²⁵

```
ConRel :  $\mathcal{V} \rightarrow \mathcal{U} \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ .$ 
ConRel I A  $\mathcal{W} = (I \rightarrow A) \rightarrow \mathcal{W} .$ 
```

We now define types that are useful for asserting and proving facts about *compatibility* of functions with continuous relations.

```
module _ { $\mathcal{U} \mathcal{V} \mathcal{W} : \text{Universe}$ } {I J :  $\mathcal{V}$ } {A :  $\mathcal{U}$ } where

lift-con-rel : ConRel I A  $\mathcal{W} \rightarrow (I \rightarrow J \rightarrow A) \rightarrow \mathcal{V} \sqcup \mathcal{W} .$ 
lift-con-rel R  $\mathfrak{a} = \forall (j : J) \rightarrow R \lambda i \rightarrow (\mathfrak{a} i) j$ 

con-compatible-fun : (I  $\rightarrow (J \rightarrow A) \rightarrow A) \rightarrow \text{ConRel} I A \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} .$ 
con-compatible-fun  $\mathbb{f} R = \forall \mathfrak{a} \rightarrow (\text{lift-con-rel } R) \mathfrak{a} \rightarrow R \lambda i \rightarrow (\mathbb{f} i) (\mathfrak{a} i)$ 
```

In the definition of *con-compatible-fun*, we let Agda infer the type of \mathfrak{a} , which is $I \rightarrow (J \rightarrow A)$.

3.2.2 Dependent relations

In this section we exploit the power of dependent types to define a completely general relation type. Specifically, we let the tuples inhabit a dependent function type, where the codomain may depend upon the input coordinate $i : I$ of the domain. Heuristically, think of the inhabitants of the following type as relations from $A \ i_1$ to $A \ i_2$ to $A \ i_3$ to \dots (This is just for intuition since the domain I need not be enumerable.)

```
DepRel : (I :  $\mathcal{V}$ ) (A : I  $\rightarrow \mathcal{U}$ ) ( $\mathcal{W} : \text{Universe}$ )  $\rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ .$ 
DepRel I A  $\mathcal{W} = \Pi A \rightarrow \mathcal{W} .$ 
```

We call *DepRel* the type of *dependent relations*.

Above we saw lifts of continuous relations and what it means for such relations to be compatible with functions. We conclude this module by defining the (only slightly more complicated) lift of dependent relations, and the type that represents compatibility of a tuple of operations with a dependent relation.

```
module _ { $\mathcal{U} \mathcal{V} \mathcal{W} : \text{Universe}$ } {I J :  $\mathcal{V}$ } {A : I  $\rightarrow \mathcal{U}$ } where

lift-dep-rel : DepRel I A  $\mathcal{W} \rightarrow (\forall i \rightarrow J \rightarrow A i) \rightarrow \mathcal{V} \sqcup \mathcal{W} .$ 
lift-dep-rel R  $\mathfrak{a} = \forall (j : J) \rightarrow R (\lambda i \rightarrow (\mathfrak{a} i) j)$ 

dep-compatible-fun : ( $\forall i \rightarrow (J \rightarrow A i) \rightarrow A i$ )  $\rightarrow \text{DepRel} I A \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} .$ 
dep-compatible-fun  $\mathbb{f} R = \forall \mathfrak{a} \rightarrow (\text{lift-dep-rel } R) \mathfrak{a} \rightarrow R \lambda i \rightarrow (\mathbb{f} i) (\mathfrak{a} i)$ 
```

In the definition of *dep-compatible-fun*, we let Agda infer the type of \mathfrak{a} , which is $(i : I) \rightarrow J \rightarrow A \ i$.

²⁵For consistency and readability, throughout the UALib we treat two universe variables with special care. The first of these is \mathfrak{O} which shall be reserved for types that represent *operation symbols* (see [AlgebrasSignatures](#)). The second is \mathcal{V} which we reserve for types representing *arities* of relations or operations.

3.3 Quotients: equivalences, class representatives, quotient types

This section presents the `Relations.Quotients` module of the `AgdaUALib`, slightly abridged.²⁶

3.3.1 Properties of binary relations

Let $\mathcal{U} : \text{Universe}$ be a universe and $A : \mathcal{U} \bullet$ a type. In `RelationsDiscrete` we defined types for representing and reasoning about binary relations on A . In this module we will define types for binary relations that have special properties. The most important special properties of relations are the ones we now define.

```
module _ { $\mathcal{U} : \text{Universe}$ } where

  reflexive : { $\mathcal{R} : \text{Universe}$ }{ $X : \mathcal{U} \bullet$ } → Rel  $X \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
  reflexive  $\_ \approx \_$  =  $\forall x \rightarrow x \approx x$ 

  symmetric : { $\mathcal{R} : \text{Universe}$ }{ $X : \mathcal{U} \bullet$ } → Rel  $X \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
  symmetric  $\_ \approx \_$  =  $\forall x y \rightarrow x \approx y \rightarrow y \approx x$ 

  antisymmetric : { $\mathcal{R} : \text{Universe}$ }{ $X : \mathcal{U} \bullet$ } → Rel  $X \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
  antisymmetric  $\_ \approx \_$  =  $\forall x y \rightarrow x \approx y \rightarrow y \approx x \rightarrow x \equiv y$ 

  transitive : { $\mathcal{R} : \text{Universe}$ }{ $X : \mathcal{U} \bullet$ } → Rel  $X \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
  transitive  $\_ \approx \_$  =  $\forall x y z \rightarrow x \approx y \rightarrow y \approx z \rightarrow x \approx z$ 
```

The `Type Topology` library also defines the following *uniqueness-of-proofs* property that a binary relation may or may not possess.

```
is-subsingleton-valued : { $\mathcal{R} : \text{Universe}$ }{ $A : \mathcal{U} \bullet$ } → Rel  $A \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
is-subsingleton-valued  $\_ \approx \_$  =  $\forall x y \rightarrow \text{is-subsingleton } (x \approx y)$ 
```

Thus, if $R : \text{Rel } A \mathcal{R}$, then `is-subsingleton-valued R` is the assertion that for each pair $x y : A$ there is *at most one proof* of $R x y$.

3.3.2 Equivalence classes

A binary relation is called a *preorder* if it is reflexive and transitive. An *equivalence relation* is a symmetric preorder. Here are the types we use to represent these concepts in the `UALib`.

```
module _ { $\mathcal{U} \mathcal{R} : \text{Universe}$ } where

  is-preorder : { $X : \mathcal{U} \bullet$ } → Rel  $X \mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R} \bullet$ 
  is-preorder  $\_ \approx \_$  = is-subsingleton-valued  $\_ \approx \_$  × reflexive  $\_ \approx \_$  × transitive  $\_ \approx \_$ 

  record IsEquivalence { $A : \mathcal{U} \bullet$ } ( $\_ \approx \_$  : Rel  $A \mathcal{R}$ ) :  $\mathcal{U} \sqcup \mathcal{R} \bullet$  where
    field
      rfl : reflexive  $\_ \approx \_$ 
      sym : symmetric  $\_ \approx \_$ 
      trans : transitive  $\_ \approx \_$ 
```

²⁶For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Quotients.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Quotients.lagda>.

```

is-equivalence-relation : {X :  $\mathcal{U}$  ·} → Rel X  $\mathcal{R}$  →  $\mathcal{U} \sqcup \mathcal{R}$  ·
is-equivalence-relation _≈_ = is-preorder _≈_ × symmetric _≈_

```

An easy first example of an equivalence relation is the kernel of any function. Here is how we prove that the kernel of a function is, indeed, an equivalence relation on the domain of the function.

```

map-kernel-IsEquivalence : { $\mathcal{U} \mathcal{W}$  : Universe}{A :  $\mathcal{U}$  ·}{B :  $\mathcal{W}$  ·}
                        (f : A → B) → IsEquivalence (KER-rel{ $\mathcal{U}$ }{ $\mathcal{W}$ } f)

map-kernel-IsEquivalence { $\mathcal{U}$ }{ $\mathcal{W}$ } f =
  record { rfl = λ x → refl
        ; sym = λ x y x1 → ≡-sym{ $\mathcal{W}$ } (f x) (f y) x1
        ; trans = λ x y z x1 x2 → ≡-trans (f x) (f y) (f z) x1 x2 }

```

3.3.3 Equivalence classes

If R is an equivalence relation on A , then for each $a : A$, there is an *equivalence class* containing a , which we denote and define by $[a] R := \text{all } b : A \text{ such that } R a b$. We often refer to $[a] R$ as the *R -class containing a* .

```

module _ { $\mathcal{U} \mathcal{R}$  : Universe} where

  [ ]_ : {A :  $\mathcal{U}$  ·} → A → Rel A  $\mathcal{R}$  → Pred A  $\mathcal{R}$ 
  [ a ] R = λ x → R a x

```

So, $x \in [a] R$ if and only if $R a x$, as desired.

We define the type of all R -classes of the relation R as follows.

```

 $\mathcal{C}$  : {A :  $\mathcal{U}$  ·}{R : Rel A  $\mathcal{R}$ } → Pred A  $\mathcal{R}$  → ( $\mathcal{U} \sqcup \mathcal{R}$  +) ·
 $\mathcal{C}$  {A} {R} C =  $\Sigma$  a : A , C ≡ ([ a ] R)

```

If R is an equivalence relation on A , then the *quotient* of A modulo R is denoted by A / R and is defined to be the collection $\{[a] R \mid a : A\}$ of equivalence classes of R . There are a few ways we could define the quotient with respect to a relation, but we find the following to be the most useful.

```

_/_ : (A :  $\mathcal{U}$  ·) → Rel A  $\mathcal{R}$  →  $\mathcal{U} \sqcup (\mathcal{R}$  +) ·
A / R =  $\Sigma$  C : Pred A  $\mathcal{R}$  ,  $\mathcal{C}$  {R = R} C

```

We define the following introduction rule for an R -class with a designated representative.

```

[ ] : {A :  $\mathcal{U}$  ·} → A → {R : Rel A  $\mathcal{R}$ } → A / R
[ a ] {R} = [ a ] R , a , refl

```

If the relation is reflexive, then we have the following elimination rules.²⁷

```

/-refl : {A :  $\mathcal{U}$  ·}{a b : A}{R : Rel A  $\mathcal{R}$ } → reflexive R → [ a ] R ≡ [ b ] R → R a b

/-refl a b rfl x = cong-app-pred b (rfl b) (x-1)

```

²⁷**Unicode Hint.** Type `⌈` and `⌋` as `\cul` and `\cur` in `agda2-mode`.

```
⌈_⌋ : {A :  $\mathcal{U}$  ·} {R : Rel A  $\mathcal{R}$ } → A / R → A
```

```
⌈ a ⌋ = || a ||
```

Later we will need the following additional quotient tools.

```
module _ { $\mathcal{U}$   $\mathcal{R}$  : Universe} {A :  $\mathcal{U}$  ·} where
```

```
open IsEquivalence { $\mathcal{U}$ } { $\mathcal{R}$ }
```

```
/-subset : {a b : A} {R : Rel A  $\mathcal{R}$ } → IsEquivalence R → R a b → [ a ] R ⊆ [ b ] R
```

```
/-subset {a} {b} Req Rab {x} Rax = (trans Req) b a x (sym Req a b Rab) Rax
```

```
/-supset : {a b : A} {R : Rel A  $\mathcal{R}$ } → IsEquivalence R → R a b → [ a ] R ⊇ [ b ] R
```

```
/-supset {a} {b} Req Rab {x} Rbx = (trans Req) a b x Rab Rbx
```

```
/-≐ : {a b : A} {R : Rel A  $\mathcal{R}$ } → IsEquivalence R → R a b → [ a ] R ≐ [ b ] R
```

```
/-≐ Req Rab = /-subset Req Rab , /-supset Req Rab
```

3.4 Truncation: continuous propositions, quotient extensionality

This section presents the `Relations.Truncation` module of the `AgdaUALib`, slightly abridged.²⁸ Here we discuss *truncation* and *h-sets* (which we just call *sets*). We first give a brief discussion of standard notions of truncation, and then we describe a viewpoint which seems useful for formalizing mathematics in Agda. Readers wishing to learn more about truncation and proof-relevant mathematics should consult other sources, such as Section 34 and 35 of Martín Escardó's notes [9], or Guillaume Brunerie, *Truncations and truncated higher inductive types*, or Section 7.1 of the HoTT book [13].²⁹

3.4.1 Background and motivation³⁰

In general, we may have many inhabitants of a given type, hence (via Curry-Howard) many proofs of a given proposition. For instance, suppose we have a type X and an identity relation \equiv_x on X so that, given two inhabitants of X , say, $a b : X$, we can form the type $a \equiv_x b$. Suppose p and q inhabit the type $a \equiv_x b$; that is, p and q are proofs of $a \equiv_x b$, in which case we write $p q : a \equiv_x b$. We might then wonder whether and in what sense are the two proofs p and q the equivalent.

We are asking about an identity type on the identity type \equiv_x , and whether there is some inhabitant, say, r of this type; i.e., whether there is a proof $r : p \equiv_{x1} q$ that the proofs of $a \equiv_x b$ are the same. If such a proof exists for all $p q : a \equiv_x b$, then the proof of $a \equiv_x b$ is unique; as a property of the types X and \equiv_x , this is sometimes called *uniqueness of identity proofs*.

Now, perhaps we have two proofs, say, $r s : p \equiv_{x1} q$ that the proofs p and q are equivalent. Then of course we wonder whether $r \equiv_{x2} s$ has a proof! But at some level we may decide

²⁸For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Truncation.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Truncation.lagda>.

²⁹**Remark.** Agda now has a built in type called `Prop` which may provide some or all of what we develop in this module. (See the [Prop Section](#) of agda.readthedocs.io.) However, we don't use it anywhere in the `UALib`; it seems we get along just fine without it.

³⁰The remarks in this subsection serve to introduce novices to the basic notion of truncation. Readers already familiar with this notion may wish to skip to the next subsection.

that the potential to distinguish two proofs of an identity in a meaningful way (so-called *proof-relevance*) is not useful or desirable. At that point, say, at level k , we would be naturally inclined to assume that there is at most one proof of any identity of the form $p \equiv_{xk} q$. This is called *truncation* (at level k) (see, e.g., the Truncation section of Escardó’s notes [9]).

3.4.2 Sets

In *homotopy type theory*, a type X with an identity relation \equiv_x is called a *set* (or *0-groupoid*) if for every pair $x y : X$ there is at most one proof of $x \equiv_x y$. In other words, the type X , along with its equality type \equiv_x , form a *set* if for all $x y : X$ there is at most one proof of $x \equiv_x y$.

This notion is formalized in the *Type Topology* library using the types *is-set* which is defined using the *is-subsingleton* type that we saw earlier (§2.4) as follows.³¹

```
is-set :  $\mathcal{U} \rightarrow \mathcal{U}$ 
is-set X = (x y : X) → is-subsingleton (x ≡ y)
```

Thus, the pair (X, \equiv_x) forms a set iff it satisfies $\forall x y : X \rightarrow \text{is-subsingleton } (x \equiv_x y)$.

The function *to-Σ-≡*, which we also import, is part of Escardó’s characterization of equality in Sigma types ([9]). It is defined as follows.

```
module hide-to-Σ-≡ { $\mathcal{U} \mathcal{W} : \text{Universe}$ } where

to-Σ-≡ : {X :  $\mathcal{U}$ } {A : X →  $\mathcal{W}$ } {σ τ : Σ A}
  → Σ p : | σ | ≡ | τ | , (transport A p || σ ||) ≡ || τ ||
  → σ ≡ τ

to-Σ-≡ (refl {x = x} , refl {x = a}) = refl {x = (x , a)}
```

```
open import MGS-Embeddings using (to-Σ-≡) public
```

We will use *is-embedding*, *is-set*, and *to-Σ-≡* in the next subsection to prove that a monic function into a set is an embedding.

3.4.3 Injective functions are set embeddings

Before moving on to define propositions, we discharge an obligation mentioned but left unfulfilled in the *embeddings* section of the *PreludeInverses* module. Recall, we described and imported the *is-embedding* type, and we remarked that an embedding is not simply a monic function. However, if we assume that the codomain is truncated so as to have unique identity proofs (i.e., is a set), then we can prove that any monic function into that codomain will be an embedding. On the other hand, embeddings are always monic, so we will end up with an equivalence. To prepare for this, we define a type $_ \iff _$ with which to represent such equivalences.

```
_iff_ : { $\mathcal{U} \mathcal{W} : \text{Universe}$ } →  $\mathcal{U} \rightarrow \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W}$ 
X iff Y = (X → Y) × (Y → X)
```

```
module _ { $\mathcal{U} \mathcal{W} : \text{Universe}$ } {A :  $\mathcal{U}$ } {B :  $\mathcal{W}$ } where
```

```
monic-is-embedding|sets : (f : A → B) → is-set B → Monic f → is-embedding f
```

³¹As Escardó explains, “at this point, with the definition of these notions, we are entering the realm of univalent mathematics, but not yet needing the univalence axiom.”

```

monic-is-embedding|sets f Bset fmon b (a , fa≡b) (a' , fa'≡b) = γ
  where
    faa' : f a ≡ f a'
    faa' = ≡-Trans (f a) (f a') fa≡b (fa'≡b -1)

    aa' : a ≡ a'
    aa' = fmon a a' faa'

    A : _ → _ .
    A a = f a ≡ b

    arg1 : Σ p : (a ≡ a') , (transport A p fa≡b) ≡ fa'≡b
    arg1 = aa' , Bset (f a') b (transport A aa' fa≡b) fa'≡b

    γ : a , fa≡b ≡ a' , fa'≡b
    γ = to-Σ-≡ arg1

```

In stating the previous result, we introduce a new convention to which we hope to adhere. Whenever a result holds only for sets, we will add the special suffix `|sets`, which hopefully calls to mind the standard mathematical notation for the restriction of a function to a subset of its domain.

Embeddings are always monic, so we conclude that when a function's codomain is a set, then that function is an embedding if and only if it is monic.

```

embedding-iff-monic|sets : (f : A → B) → is-set B
  -----
  → is-embedding f ⇔ Monic f

embedding-iff-monic|sets f Bset = (embedding-is-monic f) , (monic-is-embedding|sets f Bset)

```

3.4.4 Propositions

Sometimes we will want to assume that a type X is a *set*. As we just learned, this means there is at most one proof that two inhabitants of X are the same. Analogously, for predicates on X , we may wish to assume that there is at most one proof that an inhabitant of X satisfies the given predicate. If a unary predicate satisfies this condition, then we call it a (unary) *proposition*. We now define a type that captures this concept.

```

module _ {U : Universe} where

  Pred1 : U → (W : Universe) → U ⊔ W+ .
  Pred1 A W = Σ P : (Pred A W) , ∀ x → is-singleton (P x)

```

(Recall that $\text{Pred } A \mathcal{W}$ is simply the function type $A \rightarrow \mathcal{W}$.)

The principle of *proposition extensionality* asserts that logically equivalent propositions are equivalent. That is, if we have $P \ Q : \text{Pred}_1$ and $| P | \subseteq | Q |$ and $| Q | \subseteq | P |$, then $P \equiv Q$. This is formalized as follows (cf. the section “Prop extensionality and the powerset” of [9]).

```

prop-ext : (U W : Universe) → (U ⊔ W)+ .
prop-ext U W = ∀ {A : U} {P Q : Pred1 A W} → | P | ⊆ | Q | → | Q | ⊆ | P | → P ≡ Q

```

Recall, we defined the relation $\underline{\dot{=}}$ for predicates as follows: $P \dot{=} Q = (P \subseteq Q) \times (Q \subseteq P)$. Therefore, if we assume $\text{PropExt } A \mathcal{W} \{P\}\{Q\}$ holds, then it follows that $P \equiv Q$.

```
prop-ext' : (A :  $\mathcal{U}$  ·) ( $\mathcal{W}$  : Universe) {P Q : Pred1 A  $\mathcal{W}$ } → prop-ext A  $\mathcal{W}$ 
→ | P |  $\dot{=}$  | Q | → P  $\equiv$  Q
```

```
prop-ext' A  $\mathcal{W}$  pe hyp = pe (fst hyp) (snd hyp)
```

Thus, for truncated predicates P and Q , if `PropExt` holds, then $P \subseteq Q \times Q \subseteq P \rightarrow P \equiv Q$, which is a useful extensionality principle.

3.4.5 Binary propositions

Given a binary relation R , it may be necessary or desirable to assume that there is at most one way to prove that a given pair of elements is R -related. If this is true of R , then we call R a *binary proposition*.³²

As above, we use the `is-subsingleton` type of the `Type Topology` library to impose this truncation assumption on a binary relation.³³

```
Pred2 :  $\mathcal{U}$  · → ( $\mathcal{W}$  : Universe) →  $\mathcal{U} \sqcup \mathcal{W}^+ \cdot$ 
Pred2 A  $\mathcal{W}$  =  $\Sigma$  R : (Rel A  $\mathcal{W}$ ) ,  $\forall x y \rightarrow$  is-subsingleton (R x y)
```

(Recall, `Rel A \mathcal{W}` is simply the function type $A \rightarrow A \rightarrow \mathcal{W} \cdot$.)

3.4.6 Quotient extensionality

We need a (subsingleton) identity type for congruence classes over sets so that we can equate two classes even when they are presented using different representatives. Proposition extensionality is precisely what we need to accomplish this. (Notice that we don't require *function extensionality* (§2.3) here.)

```
module _ { $\mathcal{U}$   $\mathcal{R}$  : Universe} {A :  $\mathcal{U}$  ·} {R : Pred2 A  $\mathcal{R}$ } where

class-extensionality : prop-ext  $\mathcal{U}$   $\mathcal{R}$  → {u v : A} → lsEquivalence | R |
→ | R | u v → [ u ] | R |  $\equiv$  [ v ] | R |

class-extensionality pe {u}{v} Reqv Ruv =  $\gamma$ 
where
  P Q : Pred1 A  $\mathcal{R}$ 
  P = ( $\lambda a \rightarrow$  | R | u a) , ( $\lambda a \rightarrow$  || R || u a)
  Q = ( $\lambda a \rightarrow$  | R | v a) , ( $\lambda a \rightarrow$  || R || v a)

   $\alpha$  : [ u ] | R |  $\subseteq$  [ v ] | R |
   $\alpha$  ua = fst (/  $\dot{=}$  Reqv Ruv) ua

   $\beta$  : [ v ] | R |  $\subseteq$  [ u ] | R |
   $\beta$  va = snd (/  $\dot{=}$  Reqv Ruv) va

  PQ : P  $\equiv$  Q
  PQ = (prop-ext' pe ( $\alpha$  ,  $\beta$ ))
```

³²This is another example of *proof-irrelevance* since, if R is a binary proposition and we have two proofs of $R x y$, then we can assume that the proofs are indistinguishable or that any distinctions are irrelevant.

³³Using the definition `is-subsingleton-valued` from § 3.3.1, we could have defined `Pred2` by $\Sigma R : (\text{Rel } A \mathcal{W})$, `is-subsingleton-valued` R , but this seems less transparent than our explicit definition.

$$\begin{aligned}
& \gamma : [u] \mid \mathbf{R} \mid \equiv [v] \mid \mathbf{R} \mid \\
& \gamma = \text{ap fst PQ} \\
\\
& \text{to-subtype-} \llbracket \cdot \rrbracket : \{C D : \text{Pred } A \mathcal{R}\} \{c : \mathcal{C} C\} \{d : \mathcal{C} D\} \\
& \rightarrow (\forall C \rightarrow \text{is-subsingleton } (\mathcal{C}\{R = \mid \mathbf{R} \mid\} C)) \\
& \rightarrow C \equiv D \rightarrow (C, c) \equiv (D, d) \\
\\
& \text{to-subtype-} \llbracket \cdot \rrbracket \{D = D\} \{c\} \{d\} \text{ ssA } CD = \text{to-}\Sigma\text{-}\equiv (CD, \text{ssA } D (\text{transport } \mathcal{C} CD c) d) \\
\\
& \text{class-extensionality}' : \text{prop-ext } \mathcal{U} \mathcal{R} \rightarrow \{u v : A\} \rightarrow (\forall C \rightarrow \text{is-subsingleton } (\mathcal{C} C)) \\
& \rightarrow \text{IsEquivalence } \mid \mathbf{R} \mid \rightarrow \mid \mathbf{R} \mid u v \rightarrow \llbracket u \rrbracket \equiv \llbracket v \rrbracket \\
\\
& \text{class-extensionality}' \text{ pe } \{u\} \{v\} \text{ ssA } \text{Reqv } Ru v = \gamma \\
& \text{where} \\
& \quad CD : [u] \mid \mathbf{R} \mid \equiv [v] \mid \mathbf{R} \mid \\
& \quad CD = \text{class-extensionality } \text{pe } \text{Reqv } Ru v \\
\\
& \gamma : \llbracket u \rrbracket \equiv \llbracket v \rrbracket \\
& \gamma = \text{to-subtype-} \llbracket \cdot \rrbracket \text{ ssA } CD
\end{aligned}$$

3.4.7 Continuous proposition types

We defined a type called `ConRel` in the `RelationsContinuous` module to represent relations of arbitrary arity. Naturally, we define a type of *truncated continuous relations*, the inhabitants of which we will call *continuous propositions*.

$$\begin{aligned}
& \text{ConProp} : \mathcal{V} \cdot \rightarrow \mathcal{U} \cdot \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot \\
& \text{ConProp } I A \mathcal{W} = \Sigma P : (\text{ConRel } I A \mathcal{W}), \forall a \rightarrow \text{is-subsingleton } (P a) \\
\\
& \text{con-prop-ext} : \mathcal{V} \cdot \rightarrow \mathcal{U} \cdot \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot \\
& \text{con-prop-ext } I A \mathcal{W} = \{P Q : \text{ConProp } I A \mathcal{W}\} \rightarrow \mid P \mid \subseteq \mid Q \mid \rightarrow \mid Q \mid \subseteq \mid P \mid \rightarrow P \equiv Q
\end{aligned}$$

The point of this is that if we assume `con-prop-ext` $I A \mathcal{W}$ holds for some I, A and \mathcal{W} , then we can prove that logically equivalent continuous propositions of type `ConProp` $I A \mathcal{W}$ are equivalent.

$$\begin{aligned}
& \text{con-prop-ext}' : (I : \mathcal{V} \cdot)(A : \mathcal{U} \cdot)(\mathcal{W} : \text{Universe}) \{P Q : \text{ConProp } I A \mathcal{W}\} \\
& \rightarrow \text{con-prop-ext } I A \mathcal{W} \\
& \rightarrow \mid P \mid \dot{=} \mid Q \mid \rightarrow P \equiv Q \\
\\
& \text{con-prop-ext}' I A \mathcal{W} \text{ pe hyp} = \text{pe } \mid \text{hyp} \mid \parallel \text{hyp} \parallel
\end{aligned}$$

While we're at it, we might as well take the abstraction one step further and define *truncated dependent relations*, which we'll call *dependent propositions*.

$$\begin{aligned}
& \text{DepProp} : (I : \mathcal{V} \cdot)(A : I \rightarrow \mathcal{U} \cdot)(\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot \\
& \text{DepProp } I A \mathcal{W} = \Sigma P : (\text{DepRel } I A \mathcal{W}), \forall a \rightarrow \text{is-subsingleton } (P a) \\
\\
& \text{dep-prop-ext} : (I : \mathcal{V} \cdot)(A : I \rightarrow \mathcal{U} \cdot)(\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot
\end{aligned}$$

`dep-prop-ext` $I A \mathcal{W} = \{P Q : \text{DepProp } I A \mathcal{W}\} \rightarrow |P| \subseteq |Q| \rightarrow |Q| \subseteq |P| \rightarrow P \equiv Q$

Applying the extensionality principle for dependent relations is no harder than applying the special cases of this principle defined earlier.

```

dep-prop-ext' : (I :  $\mathcal{V}$  ·)(A :  $I \rightarrow \mathcal{U}$  ·)( $\mathcal{W} : \text{Universe}$ )
               {P Q : DepProp I A  $\mathcal{W}$ } → dep-prop-ext I A  $\mathcal{W}$ 
               -----
               → |P|  $\dot{=}$  |Q| → P  $\equiv$  Q

dep-prop-ext' I A  $\mathcal{W}$  pe hyp = pe | hyp | || hyp ||

```

4 Algebra Types

A standard way to define algebraic structures in type theory is using record types. However, we feel the dependent pair (or Sigma) type (§2.1.2) is more natural, as it corresponds semantically to the existential quantifier of logic. Therefore, many of the important types of the UALib are defined as Sigma types. In this section, we use function types and Sigma types to define the types of *operations* and *signatures* (§4.1), *algebras* (§4.2), and *product algebras* (§4.3), *congruence relations* 4.4, and *quotient algebras* 4.4.2.

4.1 Signatures: types for operations and signatures

This section presents the `Algebras.Signatures` module of the `AgdaUALib`, slightly abridged.³⁴

4.1.1 Operation type

We begin by defining the type of *operations*, and give an example (the projections).

```

module _ { $\mathcal{U} : \text{Universe}$ } where

--The type of operations
Op :  $\mathcal{V}$  · →  $\mathcal{U}$  · →  $\mathcal{U} \sqcup \mathcal{V}$  ·
Op I A = (I → A) → A

--Example. the projections
 $\pi : \{I : \mathcal{V} \cdot\} \{A : \mathcal{U} \cdot\} \rightarrow I \rightarrow \text{Op } I A$ 
 $\pi \ i \ x = x \ i$ 

```

The type `Op` encodes the arity of an operation as an arbitrary type $I : \mathcal{V}$, which gives us a very general way to represent an operation as a function type with domain $I \rightarrow A$ (the type of “tuples”) and codomain A . The last two lines of the code block above codify the i -th I -ary projection operation on A .

³⁴For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Signatures.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Signatures.lagda>.

4.1.2 Signature type

We define the signature of an algebraic structure in Agda like this.

```
Signature : (ℳ ℳ : Universe) → (ℳ ⊔ ℳ) + ·
Signature ℳ ℳ = Σ F : ℳ · , (F → ℳ ·)
```

As mentioned in the section on [Relations of arbitrary arity](#) in the [Relations.Continuous](#) module, \mathcal{M} will always denote the universe of *operation symbol* types, while \mathcal{V} is the universe of *arity* types.

In the [Prelude](#) module we defined special syntax for the first and second projections—namely, $| _ |$ and $\| _ \|$, respectively. Consequently, if $\{S : \text{Signature } \mathcal{M} \mathcal{V}\}$ is a signature, then $| S |$ denotes the set of operation symbols, and $\| S \|$ denotes the arity function. If $f : | S |$ is an operation symbol in the signature S , then $\| S \| f$ is the arity of f .

4.1.2.1 Example

Here is how we might define the signature for monoids as a member of the type $\text{Signature } \mathcal{M} \mathcal{V}$.

```
data monoid-op : ℳ · where
  e : monoid-op
  · : monoid-op

monoid-sig : Signature ℳ ℳ0
monoid-sig = monoid-op , λ { e → 0; · → 2 }
```

As expected, the signature for a monoid consists of two operation symbols, e and \cdot , and a function $\lambda \{ e \rightarrow 0; \cdot \rightarrow 2 \}$ which maps e to the empty type 0 (since e is the nullary identity) and maps \cdot to the two element type 2 (since \cdot is binary).³⁵

4.2 Algebras: types for algebras, operation interpretation, and compatibility

This section presents the [Algebras.Algebras](#) module of the [AgdaUALib](#), slightly abridged.³⁶

4.2.1 The Algebra type

For a fixed signature $S : \text{Signature } \mathcal{M} \mathcal{V}$ and universe \mathcal{U} , we define the type of *algebras in the signature S* (or *S -algebras*) and with *domain* (or *carrier* or *universe*) $A : \mathcal{U}$ as follows.

```
Algebra : (ℳ : Universe)(S : Signature ℳ ℳ) → ℳ ⊔ ℳ ⊔ ℳ + ·
Algebra ℳ S = Σ A : ℳ · , ((f : | S |) → Op (‖ S ‖ f) A)
```

We could refer to an inhabitant of this type as a “ ∞ -algebra” because its domain can be an arbitrary type, say, $A : \mathcal{U}$ and need not be truncated at some level; in particular, A need not be a *set*. (See the discussion in §3.4.2.)

We might take this opportunity to define the type of “0-algebras” (algebras whose domains are sets), which is probably closer to what most of us think of when doing informal universal

³⁵The types 0 and 2 are defined in the [MGS-MLTT](#) module of the [Type Topology](#) library.

³⁶For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Algebras.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Algebras.lagda>.

algebra. However, below we will only need to know that the domains of our algebras are sets in a few places in the `UALib`, so it seems preferable to work with general (∞ -)algebras throughout and then assume uniqueness of identity proofs explicitly and only where needed.

4.2.2 Operation interpretation syntax

We now define a convenient shorthand for the interpretation of an operation symbol. This looks more similar to the standard notation one finds in the literature as compared to the double bar notation we started with, so we will use this new notation almost exclusively in the remaining modules of the `UALib`.

$$\begin{aligned} _ \hat{_} &: (f : | S |)(\mathbf{A} : \text{Algebra } \mathcal{U} S) \rightarrow (|| S || f \rightarrow | \mathbf{A} |) \rightarrow | \mathbf{A} | \\ f \hat{\mathbf{A}} &= \lambda a \rightarrow (|| \mathbf{A} || f) a \end{aligned}$$

So, if $f : | S |$ is an operation symbol in the signature S , and if $a : || S || f \rightarrow | \mathbf{A} |$ is a tuple of the appropriate arity, then $(f \hat{\mathbf{A}}) a$ denotes the operation f interpreted in \mathbf{A} and evaluated at a .

4.2.3 Arbitrarily many variable symbols

We sometimes want to assume that we have at our disposal an arbitrary collection X of variable symbols such that, for every algebra \mathbf{A} , no matter the type of its domain, we have a surjective map $h : X \rightarrow | \mathbf{A} |$ from variables onto the domain of \mathbf{A} . We may use the following definition to express this assumption when we need it.

$$\begin{aligned} _ \twoheadrightarrow _ &: \{S : \text{Signature } \mathcal{O} \mathcal{V}\} \{ \mathcal{U} \mathcal{X} : \text{Universe} \} \rightarrow \mathcal{X} \rightarrow \text{Algebra } \mathcal{U} S \rightarrow \mathcal{X} \sqcup \mathcal{U} \cdot \\ X \twoheadrightarrow \mathbf{A} &= \Sigma h : (X \rightarrow | \mathbf{A} |) , \text{Epic } h \end{aligned}$$

Now we can assert, in a specific module, the existence of the surjective map described above by including the following line in that module's declaration, like so.

```
module _ {X : {U X : Universe} {X : X} (A : Algebra U S) → X → A} where
```

Then `fst(X A)` will denote the surjective map $h : X \rightarrow | \mathbf{A} |$, and `snd(X A)` will be a proof that h is surjective.

4.2.4 Lifts of algebras

Here we define some domain-specific lifting tools for our operation and algebra types.

```
module _ {O V : Universe} {S : Signature O V} where
  Σ F : O · , ( F → V · ) } where

  lift-op : {U : Universe} {I : V ·} {A : U ·} → ((I → A) → A) → (W : Universe)
    → ((I → Lift {W} A) → Lift {W} A)

  lift-op f W = λ x → lift (f (λ i → Lift.lower (x i)))

  open algebra

  lift-alg : {U : Universe} → Algebra U S → (W : Universe) → Algebra (U ⊔ W) S
  lift-alg A W = Lift | A | , (λ (f : | S |) → lift-op (f ^ A) W)
```

```
lift-alg-record-type : {U : Universe} → algebra U S → (W : Universe) → algebra (U ⊔ W) S
lift-alg-record-type A W = mkalg (Lift (univ A)) (λ (f : | S |) → lift-op ((op A) f) W)
```

We use the function `lift-alg` to resolve errors that arise when working in Agda's noncumulative hierarchy of type universes. (See the discussion in `Prelude.Lifts`.)

4.2.5 Compatibility of binary relations

If \mathbf{A} is an algebra and R a binary relation, then `compatible A R` will represent the assertion that R is *compatible* with all basic operations of \mathbf{A} . Recall, informally this means for every operation symbol $f : | S |$ and all pairs $a \ a' : || S || f \rightarrow | \mathbf{A} |$ of tuples from the domain of \mathbf{A} , the following implication holds:

if $R (a \ i) (a' \ i)$ for all i , then $R ((f \ \hat{\ } \mathbf{A}) a) ((f \ \hat{\ } \mathbf{A}) a')$.

The formal definition representing this notion of compatibility is easy to write down since we already have a type that does all the work.

```
module _ {U W : Universe} {S : Signature ⊞ V} where
  compatible : (A : Algebra U S) → Rel | A | W → ⊞ ⊔ U ⊔ V ⊔ W ·
  compatible A R = ∀ f → compatible-fun (f ^ A) R
```

Recall the `compatible-fun` type was defined in `Relations.Discrete` module.

4.2.6 Compatibility of continuous relations

Next we define a type that represents *compatibility of a continuous relation* with all operations of an algebra. First, we define compatibility with a single operation.

```
module _ {U W : Universe} {S : Signature ⊞ V} {A : Algebra U S} {I : V ·} where
  con-compatible-op : | S | → ConRel I | A | W → U ⊔ V ⊔ W ·
  con-compatible-op f R = con-compatible-fun (λ _ → (f ^ A)) R
```

In case it helps the reader understand `con-compatible-op`, we redefine it explicitly without the help of `con-compatible-fun`.

```
con-compatible-op' : | S | → ConRel I | A | W → U ⊔ V ⊔ W ·
con-compatible-op' f R = ∀ a → (lift-con-rel R) a → R (λ i → (f ^ A) (a i))
```

where we have let Agda infer the type of \mathfrak{a} , which is $(i : I) \rightarrow || S || f \rightarrow | \mathbf{A} |$.

With `con-compatible-op` in hand, it is a trivial matter to define a type that represents *compatibility of a continuous relation with an algebra*.

```
con-compatible : ConRel I | A | W → ⊞ ⊔ U ⊔ V ⊔ W ·
con-compatible R = ∀ (f : | S |) → con-compatible-op f R
```

4.3 Products: types for products over arbitrary classes of algebras

This section presents the `Algebras.Products` module of the `AgdaUAlib`, slightly abridged.³⁷ We begin this module by assuming a signature $S : \text{Signature } \otimes V$ which is then present and available throughout the module. Because of this, in contrast to our (highly abridged) descriptions

³⁷For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Products.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UAlib/Algebras/Products.lagda>.

of previous modules, we present the first few lines of the `Algebras.Products` module in full. They are as follows.

```
{-# OPTIONS -without-K -exact-split -safe #-}
open import Algebras.Signatures using (Signature; Ⓞ; ℳ)
module Algebras.Products {S : Signature Ⓞ ℳ} where
open import Algebras.Algebras hiding (Ⓞ; ℳ) public
```

Notice that we import the `Signature` type from the `Algebras.Signatures` module first, before the `module` line, so that we may use it to declare the signature S as a parameter of the `Algebras.Products` module.

The product of S -algebras is defined as follows.

```
⊠ : {ℳ : Universe} {I : ℳ → *} (A : I → Algebra ℳ S) → Algebra (ℳ ⊔ ℳ) S

⊠ A = (∀ i → | A i |) ,                                - domain of the product algebra

λ f a i → (f ^ A i) λ x → a x i                       - basic operations of the product algebra
```

4.3.1 Products of classes of algebras

An arbitrary class \mathcal{K} of algebras is represented as a predicate over the type `Algebra ℳ S`, for some universe \mathcal{U} and signature S . That is, $\mathcal{K} : \text{Pred (Algebra } \mathcal{U} S) _$.³⁸ Later we will formally state and prove that the product of all subalgebras of algebras in such a class belongs to `SP(ℳ)` (subalgebras of products of algebras in \mathcal{K}). That is, $\prod S(\mathcal{K}) \in \text{SP}(\mathcal{K})$. This turns out to be a nontrivial exercise. In fact, it is not even clear (at least not to this author) how one should express the product of an entire class of algebras as a dependent type. However, if one ponders this for a while, the right type will eventually reveal itself, and will then seem obvious.³⁹ The solution is the `class-product` type whose construction is the main goal of this section.

First, we need a type that will serve to index the class, as well as the product of its members.⁴⁰

```
module _ {ℳ : Universe} {X : ℳ → *} where

J : Pred (Algebra ℳ S)(ov ℳ) → (X ⊔ ov ℳ) → *

J K = Σ A : (Algebra ℳ S) , (A ∈ K) × (X → | A |)
```

Notice that the second component of this dependent pair type is $(A \in \mathcal{K}) \times (X \rightarrow | A |)$. In previous versions of the `UALib` this second component was simply $A \in \mathcal{K}$, until we realized that adding the type $X \rightarrow | A |$ is quite useful. Later we will see exactly why, but for now suffice it to say that a map of type $X \rightarrow | A |$ may be viewed abstractly as an *ambient context*, or more concretely, as an assignment of *values* in $| A |$ to *variable symbols* in X . When computing with or reasoning about products, while we don't want to rigidly impose a context in advance, want

³⁸The underscore is merely a placeholder for the universe of the predicate type and doesn't concern us here.

³⁹At least this was our experience, but readers are encouraged to try to come up with a type that represents the product of all members of an inhabitant of a predicate over `Algebra ℳ S`, or even an arbitrary predicate.

⁴⁰**Notation.** Given a signature $S : \text{Signature } \mathbb{O} \mathcal{V}$, the type `Algebra ℳ S` has universe $\mathbb{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$. In the `UALib`, such universes abound, and \mathbb{O} and \mathcal{V} remain fixed throughout the library. So, for notational convenience, we define the following shorthand for universes of this form: $\text{ov } \mathcal{U} = \mathbb{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$

do want to lay our hands on whatever context is ultimately assumed. Including the “context map” inside the index type \mathcal{J} of the product turns out to be a convenient way to achieve this flexibility.

Taking the product over the index type \mathcal{J} requires a function that maps an index $i : \mathcal{J}$ to the corresponding algebra. Each index $i : \mathcal{J}$ denotes a triple, say, (\mathbf{A} , p , h) , where

$$\mathbf{A} : \text{Algebra } \mathcal{U} S, \quad p : \mathbf{A} \in \mathcal{K}, \quad h : X \rightarrow |\mathbf{A}|,$$

so the function mapping an index to the corresponding algebra is simply the first projection.

$$\mathfrak{A} : (\mathcal{K} : \text{Pred } (\text{Algebra } \mathcal{U} S)(\text{ov } \mathcal{U})) \rightarrow \mathcal{J} \mathcal{K} \rightarrow \text{Algebra } \mathcal{U} S$$

$$\mathfrak{A} \mathcal{K} = \lambda (i : (\mathcal{J} \mathcal{K})) \rightarrow |i|$$

Finally, we define `class-product` which represents the product of all members of \mathcal{K} .

$$\text{class-product} : \text{Pred } (\text{Algebra } \mathcal{U} S)(\text{ov } \mathcal{U}) \rightarrow \text{Algebra } (X \sqcup \text{ov } \mathcal{U}) S$$

$$\text{class-product } \mathcal{K} = \prod (\mathfrak{A} \mathcal{K})$$

If $p : \mathbf{A} \in \mathcal{K}$ and $h : X \rightarrow |\mathbf{A}|$, then we can think of the triple $(\mathbf{A} , p , h) \in \mathcal{J} \mathcal{K}$ as an index over the class, and so we can think of $\mathfrak{A} (\mathbf{A} , p , h)$ (which is simply \mathbf{A}) as the projection of the product $\prod (\mathfrak{A} \mathcal{K})$ onto the (\mathbf{A} , p , h) -th component.

4.4 Congruences: types for congruences and quotient algebras

This section presents the `Algebras.Congruences` module of the `AgdaUALib`, slightly abridged.⁴¹ A *congruence relation* of an algebra \mathbf{A} is defined to be an equivalence relation that is compatible with the basic operations of \mathbf{A} . This concept can be represented in a number of different ways in type theory. For example, we define both a Sigma type `Con` and a record type `Congruence`, each of which captures the informal notion of congruence, and each one is useful in certain contexts. (We will see examples later.)

```
Con : {U : Universe} {A : Algebra U S} → ov U ·
Con {U} A = Σ θ : ( Rel | A | U ) , IsEquivalence θ × compatible A θ

record Congruence {U W : Universe} (A : Algebra U S) : ov W ⊔ U · where
  constructor mkcon
  field
    ⟨_⟩ : Rel | A | W
    Compatible : compatible A ⟨_⟩
    IsEquiv : IsEquivalence ⟨_⟩

open Congruence
```

4.4.1 Example

We defined the zero relation `0-rel` in the `Relations.Discrete` module, and we now demonstrate how to build the trivial congruence out of this relation.

⁴¹ For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Congruences.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Congruences.lagda>.

The relation `0-rel` is equivalent to the identity relation `≡` and these are obviously both equivalences. In fact, we already proved this of `≡` in the `Prelude.Equality` module, so we simply apply the corresponding proofs.

```
module _ {U : Universe} where

  0-IsEquivalence : {A : U → U} → IsEquivalence {U} {A = A} 0-rel
  0-IsEquivalence = record { rfl = ≡-rfl; sym = ≡-sym; trans = ≡-trans }
```

Next we formally record another obvious fact—namely, that `0-rel` is compatible with all operations of all algebras.

```
module _ {U : Universe} where

  0-compatible-op : funext ∀ U → {A : Algebra U S} (f : | S |) → compatible-fun (f ^ A) 0-rel
  0-compatible-op fe {A} f ptws0 = ap (f ^ A) (fe (λ x → ptws0 x))

  0-compatible : funext ∀ U → {A : Algebra U S} → compatible A 0-rel
  0-compatible fe {A} = λ f args → 0-compatible-op fe {A} f args
```

Finally, we have the ingredients need to construct the zero congruence of any algebra we like.

```
Δ : {U : Universe} → funext ∀ U → {A : Algebra U S} → Congruence A
Δ fe = mkcon 0-rel (0-compatible fe) 0-IsEquivalence
```

4.4.2 Quotient Algebras

An important construction in universal algebra is the quotient of an algebra \mathbf{A} with respect to a congruence relation θ of \mathbf{A} . This quotient is typically denote by \mathbf{A} / θ and Agda allows us to define and express quotients using the standard notation.⁴²

```
—/— : {U R : Universe} {A : Algebra U S} → Congruence {U} {R} A → Algebra (U ⊔ R+) S

A / θ = ( | A | / ⟨ θ ⟩ ) , - the domain of the quotient algebra

λ f a → [ (f ^ A) (λ i → [ a i ]) ] - the basic operations of the quotient algebra
```

4.4.3 Examples

The zero element of a quotient can be expressed as follows.

```
module _ {U R : Universe} where

  Zero/ : {A : Algebra U S} {θ : Congruence {U} {R} A} → Rel (| A | / ⟨ θ ⟩) (U ⊔ R+)

  Zero/ θ = λ x x1 → x ≡ x1
```

Finally, the following elimination rule is sometimes useful.

```
—/-refl : {A : Algebra U S} {θ : Congruence {U} {R} A} {a a' : | A |}
→ [ a ] / ⟨ θ ⟩ ≡ [ a' ] / ⟨ θ ⟩ a a'

—/-refl θ refl = IsEquivalence.rfl (IsEquiv θ) —
```

⁴²Unicode Hints. Produce the `/` symbol in `agda2-mode` by typing `\---` and then `C-f` a number of times.

5 Concluding Remarks

We’ve reached the end of Part 1 of our three-part series describing the [AgdaUALib](#). Part 2 will cover homomorphism, terms, and subalgebras, and Part 3 will cover free algebras, equational classes of algebras (i.e., varieties), and Birkhoff’s HSP theorem.

We conclude by noting that one of our goals is to make computer formalization of mathematics more accessible to mathematicians working in universal algebra and model theory. We welcome feedback from the community and are happy to field questions about the [UALib](#), how it is installed, and how it can be used to prove theorems that are not yet part of the library. Merge requests submitted to the UALib’s main gitlab repository are especially welcomed. Please visit the repository at <https://gitlab.com/ualib/ualib.gitlab.io/> and help us improve it.

References

- 1 Clifford Bergman. *Universal Algebra: fundamentals and selected topics*, volume 301 of *Pure and Applied Mathematics (Boca Raton)*. CRC Press, Boca Raton, FL, 2012.
- 2 Ana Bove and Peter Dybjer. *Dependent Types at Work*, pages 57–99. Springer Berlin Heidelberg, Berlin, Heidelberg, 2009. doi:10.1007/978-3-642-03153-3_2.
- 3 Venanzio Capretta. Universal algebra in type theory. In *Theorem proving in higher order logics (Nice, 1999)*, volume 1690 of *Lecture Notes in Comput. Sci.*, pages 131–148. Springer, Berlin, 1999. URL: http://dx.doi.org/10.1007/3-540-48256-3_10, doi:10.1007/3-540-48256-3_10.
- 4 Jesper Carlström. A constructive version of birkhoff’s theorem. *Mathematical Logic Quarterly*, 54(1):27–34, 2008. URL: <https://onlinelibrary.wiley.com/doi/abs/10.1002/malq.200710023>, arXiv:<https://onlinelibrary.wiley.com/doi/pdf/10.1002/malq.200710023>, doi:<https://doi.org/10.1002/malq.200710023>.
- 5 Alonzo Church. A formulation of the simple theory of types. *The Journal of Symbolic Logic*, 5(2):56–68, 1940. URL: <http://www.jstor.org/stable/2266170>.
- 6 William DeMeo. The Agda Universal Algebra Library and Birkhoff’s Theorem in Dependent Type Theory. *CoRR*, abs/2101.10166, 2021. source code: <https://gitlab.com/ualib/ualib.gitlab.io>. URL: <https://arxiv.org/abs/2101.10166>, arXiv:2101.10166.
- 7 William DeMeo. The Agda Universal Algebra Library, Part 2: Structure. *CoRR*, abs/2103.09092, 2021. Source code: <https://gitlab.com/ualib/ualib.gitlab.io>. URL: <https://arxiv.org/abs/2103.09092>, arXiv:2103.09092.
- 8 William DeMeo. The Agda Universal Algebra Library, Part 3: Identity. *CoRR*, 2021. (to appear) Source code: <https://gitlab.com/ualib/ualib.gitlab.io>. URL: http://arxiv.org/a/demeo_w_1.
- 9 Martín Hötzel Escardó. Introduction to univalent foundations of mathematics with Agda. *CoRR*, abs/1911.00580, 2019. URL: <http://arxiv.org/abs/1911.00580>, arXiv:1911.00580.
- 10 Emmanuel Gunther, Alejandro Gadea, and Miguel Pagano. Formalization of universal algebra in Agda. *Electronic Notes in Theoretical Computer Science*, 338:147 – 166, 2018. The 12th Workshop on Logical and Semantic Frameworks, with Applications (LSFA 2017). URL: <http://www.sciencedirect.com/science/article/pii/S1571066118300768>, doi:<https://doi.org/10.1016/j.entcs.2018.10.010>.
- 11 Ulf Norell. Dependently typed programming in agda. In *International school on advanced functional programming*, pages 230–266. Springer, 2008.
- 12 Ulf Norell. Dependently typed programming in Agda. In *Proceedings of the 6th International Conference on Advanced Functional Programming*, AFP’08, pages 230–266, Berlin, Heidelberg, 2009. Springer-Verlag. URL: <http://dl.acm.org/citation.cfm?id=1813347.1813352>.
- 13 The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Lulu and The Univalent Foundations Program, Institute for Advanced Study, 2013. URL: <https://homotopytypetheory.org/book>.

- 14 Bas Spitters and Eelis van der Weegen. Type classes for mathematics in type theory. *CoRR*, abs/1102.1323, 2011. URL: <http://arxiv.org/abs/1102.1323>, [arXiv:1102.1323](https://arxiv.org/abs/1102.1323).
- 15 The Agda Team. Agda Language Reference section on Axiom K, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/without-k.html>.
- 16 The Agda Team. Agda Language Reference section on Safe Agda, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/safe-agda.html#safe-agda>.
- 17 The Agda Team. Agda Tools Documentation section on Pattern matching and equality, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/tools/command-line-options.html#pattern-matching-and-equality>.
- 18 Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. July 2020. URL: <http://plfa.inf.ed.ac.uk/20.07/>.